Саратовский государственный университет имени Н.Г.Чернышевского
Базовая кафедра полимеров

С.Л. Шмаков

Сборник программ MAXIMA по статистической термодинамике

Саратов, 2016

УДК 531.19: 519.67

Шмаков С.Л. Сборник программ MAXIMA по статистической термодинамике: Учебно-методич. пособие. – Саратов: Саратовск. госуниверситет, 2016. – 60 с.

В пособии собраны программы (документы) пакета компьютерной математики MAXIMA, оптимизирующие решение задач статистической термодинамики. Сравниваются решения задач традиционными методами и с привлечением компьютера. Программы открыты для доработки.

Пособие предназначено для студентов Института химии СГУ, направление «Химия», профиль «Физическая химия», бакалавриат, дисциплина «Алгоритмические подходы в статистической термодинамике». Может использоваться как при проведении практических занятий в дисплейном классе, так и при чтении лекций с проектором и ноутбуком. Также будет полезно самостоятельно осваивающим термодинамику и статистическую термодинамику, магистрантам, аспирантам и молодым специалистам.

Учебно-методическое пособие рекомендовано учебно-методической комиссией Института химии Саратовского государственного университета для бакалавров, магистрантов и аспирантов, обучающихся по направлению подготовки «Химия»

Предисловие

Преподавание дисциплин на современном методическом уровне часто осложняется необходимостью проводить математические выкладки — последовательно от одной формулы к другой. При таких преобразованиях центр тяжести может незаметно сместиться на чистую математику, а физический смысл — ускользнуть. Запись формул в готовом виде чревата ошибками в их понимании и провоцирует пассивность студентов на лекциях.

В интересах решения экономических и научных задач будущих специалистов следует со студенческой скамьи приучать к работе с компьютером и математическими пакетами, в том числе для решения задач с аналитическими преобразованиями, ведь обычно применение компьютеров в вузах ограничивается решением числовых задач, лекционными презентациями и автоматизированным тестированием.

Из целого ряда универсальных математических программ нами была выбрана Maxima — свободно распространяемая программа-ядро с оболочкой для Windows — приложением wxMaxima. Интерфейс последней несколько уступает графическому интерфейсу коммерческих программ, однако это вполне компенсируется доступностью: любой студент может установить Maxima на свой компьютер, и мы рекомендуем ему это сделать.

Настоящий сборник ориентирован на курс «Алгоритмические подходы в статистической термодинамике», который автор проводит со студентами IV курса Института химии СГУ, профиль «физическая химия». Под «программой Maxima» мы понимаем совокупность последовательных ячеек (строк) с операторами, взаимосвязанных по существу и направленных на получение определённого результата, независимо от того, оформлена ли данная совокупность программным блоком формально. Создание (включая отладку) таких программ есть не что иное, как программирование, поэтому термин «документ Maxima» нами не используется: тем более что отлаженную последовательность операторов не составляет труда оформить в виде «официального» программного блока.

Существует опасность того, что студенты будут механически набирать тексты готовых программ, не вникая в их смысл и обращая внимание лишь на технические ошибки. Для профилактики этого рекомендуется с самого начала предупредить их, что после каждого набора готовой программы будет дано самостоятельное задание, которое можно выполнить лишь при условии понимания смысла набранных операторов. Кроме того, собранные здесь программы можно использовать и на лекциях, предложив студенту-добровольцу сопряжённый с проектором ноутбук.

На самом деле компьютер не может полностью заменить человека. В частности, Maxima «не понимает», в каком именно виде оператор хочет получить от неё результат. Иногда желаемой формы итогового выражения можно добиться методом проб и ошибок, перебрав разные способы упрощения выражения, но чаще всего приходится переписывать его вручную, соблюдая неписаные правила компоновки формул в физике. Этому студентов тоже надо научить.

В листингах часто встречается значок «$», означающий подавление вывода. Это делается тогда, когда вывод ячейки зеркально дублирует ввод, а также для экономии места. Студенту рекомендуется научиться распознавать эти случаи и, во втором их них, пробовать снимать этот значок. Можно также «разгружать» ячейки, разбивая их на отдельные операторы.

Надеемся, что настоящий сборник окажется полезным и тем, кто использует в своей практике другие математические пакеты, включая коммерческие; бакалаврам, магистрантам и аспирантам, специализирующимся по физической химии. Доработка приведённых в сборнике программ, создание и отладка новых будет приветствоваться и поощряться.

Автор благодарен проф. Кленину В.И. за обсуждение материалов.

1. Элементы термодинамики

Этот раздел может быть полезен в тех случаях, когда практические занятия начинаются раньше лекций. Если студенты ещё не проходили пакет Maxima, здесь можно научить их работать с ним, причём на понятном им материале.

Задача 1 (3.4 из [1]). Определить свободную энергию системы F и найти уравнение состояния, если энтропия системы определяется формулой

$$S = R\frac{V_0}{V}\left(\frac{T}{T_0}\right)^a,$$

где V_0, T_0, a — фиксированные постоянные.

Традиционное решение.

Для обратимого процесса: $dF = -pdV - SdT$, откуда

$$S = -\left(\frac{\partial F}{\partial T}\right)_V \text{ и } p = -\left(\frac{\partial F}{\partial V}\right)_T,$$

Таким образом,

$$F = -\int_{T_0}^{T} R\frac{V_0}{V}\left(\frac{T}{T_0}\right)^a dT = R\frac{V_0 T_0}{V(a+1)}\left[1-\left(\frac{T}{T_0}\right)^{a+1}\right],$$

$$p = -\left(\frac{\partial F}{\partial V}\right)_T = R\frac{V_0 T_0}{V^2(a+1)}\left[1-\left(\frac{T}{T_0}\right)^{a+1}\right].$$

Компьютерное решение.

```
(%i1) assume(notequal(a,-1))$ depends(F,[T,V])$
(%i3) S(V,T):=R*(V0/V)*(T/T0)^a$
(%i4) S(V,T)=-diff(F,T);
```

(%o4) $$\frac{R\left(\frac{T}{T0}\right)^a V0}{V} = -\frac{d}{d\,T}F$$

```
(%i5) ode2(%,F,T);
```

(%o5) $$F = \%c - \frac{R\left(\frac{T}{T0}\right)^{a+1} T0\ V0}{(a+1)\,V}$$

```
(%i6) ic1(%,T=T0,F=0),factor;
```

(%o6) $$F = \frac{R\left(T0 - T\left(\frac{T}{T0}\right)^a\right)V0}{(a+1)\,V}$$

```
(%i7) diff(rhs(%),V);
```

(%o7) $$-\frac{R\left(T0 - T\left(\frac{T}{T0}\right)^a\right)V0}{(a+1)\,V^2}$$

Задача 2 (3.14 из [1]). Теплоёмкость системы C_p и объём V выражаются уравнениями: $C_p = \alpha T^3 \ln p$, $V = \beta T^4/p$, где α и β — константы. Найти энтальпию системы.

Традиционное решение. Воспользуемся соотношением

$$C_p = \left(\frac{\partial H}{\partial T}\right)_p,$$

тогда

$$H = \int C_p dT + f(p) = \int \alpha T^3 \ln p \, dT + f(p) = \frac{\alpha T^4}{4} \ln p + f(p).$$

Из условия $(\partial H/\partial p) = V$ находим

$$\frac{\alpha T^4}{4p} + f'(p) = V = \frac{\beta T^4}{p}.$$

Поскольку $f(p)$ не зависит от T, то $\alpha/4 = \beta$ и $f'(p) = 0$, поэтому

$$H = H_0 + \frac{\alpha T^4}{4} \ln p.$$

Компьютерное решение.

```
(%i1) Cp(p,T):=alpha*T^3*log(p)$
(%i2) V(p,T):=beta*T^4/p$
(%i3) depends(H,[p,T])$ Cp(p,T)=diff(H,T);
```

$$\text{(\%o4)} \quad \alpha \log(p)\, T^3 = \frac{d}{dT} H$$

```
(%i5) ode2(%,H,T); ic1(%,T=0,H=fp);
```

$$\text{(\%o5)} \quad H = \frac{\alpha \log(p)\, T^4}{4} + \%c$$

$$\text{(\%o6)} \quad H = \frac{\alpha \log(p)\, T^4 + 4\, fp}{4}$$

```
(%i7) depends(fp,p)$ V(p,T)=diff(rhs(%th(2)),p),expand;
```

$$\text{(\%o8)} \quad \frac{\beta T^4}{p} = \frac{\alpha T^4}{4p} + \frac{d}{dp} fp$$

```
(%i9) subst([fp=H0],%th(3)),expand;
```

$$\text{(\%o9)} \quad H = \frac{\alpha \log(p)\, T^4}{4} + H0$$

Задача 3 (3.15 из [1]). Найти термодинамический потенциал Φ системы, если $C_v = \alpha V T^3$, $p = \beta T^4$.

Традиционное решение. Найдём внутреннюю энергию системы

$$E = \int_0^T C_v dT + f(V) = \frac{\alpha V T^4}{4} + f(V).$$

Так как $(\partial E/\partial V)_S = -p$, то

$$\frac{\alpha T^4}{4} + f'(V) = -p = -\beta T^4.$$

Из этого следует, что $f'(V) = 0$ и $\beta = -\alpha/4$. Тогда

$$E = E_0 + \frac{\alpha V T^4}{4}.$$

Пренебрегая нулевой энергией, получим: $E = \frac{\alpha V T^4}{4}$.

Термодинамический потенциал системы: $\Phi = E - TS + pV = F + pV$.

Свободная энергия:

$$F = E_0 - T\int_0^T \frac{dT}{T^2}\int_0^T C_v dT = E_0 - T\int_0^T \frac{\alpha V T^4}{4T^2} dT = E_0 - \frac{\alpha V T^4}{12}.$$

Тогда

$$\Phi = \Phi_0 - \frac{\alpha V T^4}{12} - \frac{\alpha V T^4}{4} = \Phi_0 - \frac{\alpha V T^4}{3}.$$

<u>Компьютерное решение</u>.

```
(%i1) depends(f,p)$
(%i2) Cv(V,T):=alpha*V*T^3$ p(T):=beta*T^4$
(%i4) E:integrate(Cv(V,T),T,0,T)+f(V);
```

(%o4) $f(V)+\frac{\alpha T^4 V}{4}$

```
(%i5) diff(E,V)=-p(T);
```

(%o5) $\frac{d}{dV} f(V)+\frac{\alpha T^4}{4}=-\beta T^4$

```
(%i6) subst([diff(f(V),V)=0],%);
```

(%o6) $\frac{\alpha T^4}{4}=-\beta T^4$

```
(%i7) subst([f(V)=0],%th(3));
```

(%o7) $\frac{\alpha T^4 V}{4}$

```
(%i8) depends(F,T)$ %th(2)=F-T*diff(F,T);
```

(%o9) $\frac{\alpha T^4 V}{4}=F-\left(\frac{d}{dT} F\right) T$

```
(%i10) ode2(%,F,T); ic1(%,V=0,F=0);
```

(%o10) $F=T\left(\%c-\frac{\alpha T^3 V}{12}\right)$

(%o11) $F=-\frac{\alpha T^4 V}{12}$

```
(%i12) rhs(%)+p(T)*V;
```

(%o12) $\beta T^4 V-\frac{\alpha T^4 V}{12}$

```
(%i13) subst(solve(%th(7),beta),%);
```

(%o13) $-\frac{\alpha T^4 V}{3}$

Задача 4 (3.16 из [1]). Найти энтальпию $H = f(p,S)$ системы, термодинамический потенциал которой

$$\Phi = \Phi_0 + AT + Bp + CT^2/2 + DTp + Ep^2/2,$$

где Φ_0 — значение термодинамического потенциала в состоянии равновесия, A, B, C, D, E — константы.

Традиционное решение. Как известно,

$$H = \Phi - T\left(\frac{\partial\Phi}{\partial T}\right).$$

Так как $(\partial\Phi/\partial T) = A + CT + Dp$, то

$$H = \Phi_0 + At + bp + \frac{CT^2}{2} + DTp + \frac{Ep^2}{2} - AT - CT^2 - DpT =$$

$$= \Phi_0 + Bp - \frac{CT^2}{2} + \frac{Ep^2}{2}.$$

Учитывая, что $(\partial\Phi/\partial T)_p = -S$, получим

$$T = -\frac{A+Dp+S}{C}.$$

Тогда $H = \Phi_0 + Bp - \frac{(A+Dp+S)^2}{2C} + \frac{Ep^2}{2}$.

Компьютерное решение.

```
(%i1) G(p,T):=G0+A*T+B*p+C*T^2/2+D*T*p+E*p^2/2$
(%i2) H:G(p,T)-T*diff(G(p,T),T),expand;
```

$$(\%o2)\quad -\frac{C\,T^2}{2}+G0+\frac{p^2\,E}{2}+p\,B$$

```
(%i3) S=-diff(G(p,T),T);
```

$$(\%o3)\quad S=-C\,T-p\,D-A$$

```
(%i4) solve(%,T);
```

$$(\%o4)\quad [T=-\frac{S+p\,D+A}{C}]$$

```
(%i5) subst(%,%th(3));
```

$$(\%o5)\quad -\frac{(S+p\,D+A)^2}{2\,C}+G0+\frac{p^2\,E}{2}+p\,B$$

Задача 5 (I-2-9/11 из [2]). Выведите барометрическую формулу $P = P_0 e^{-Mgz/RT}$, решив дифференциальное уравнение, для составления которого используйте следующее положение: изменение давления ($-dP$) между высотами z и $z + dz$ равно весу газа, приходящегося на единицу площади слоя толщиной dz. Приравняйте эти две величины, выразив их через две переменные P и z (используйте закон идеальных газов), и решите уравнение, разделив переменные.

Усложнение: ускорение силы тяжести g изменяется с увеличением расстояния r от центра Земли по формуле: $g = Gm_E/r^2$, где G — постоянная тяготения, m_E — масса Земли. Выведите видоизменение барометрической формулы, приняв во внимание это изменение g.

Второе усложнение: в нижней части атмосферы температура воздуха не одинакова, но линейно уменьшается с высотой согласно уравнению $T = T_0 - az$, где a — коэффициент пропорциональности, z — высота, T_0 — температура на уровне Земли и T — температура на высоте z. Выведите видоизменение барометрического уравнения, учитывающее эту зависимость.

<u>Традиционное решение</u>.

$$-dP = \rho g dz = \frac{Pmg}{RT} dz \text{ (}\rho\text{ — плотность);}$$

$$\frac{dP}{P} = -\frac{Mg}{RT} dz;\ \ln\frac{P}{P_0} = -\frac{Mgz}{RT},\ \ P = P_0 e^{-Mgz/RT}.$$

$$\frac{dP}{P} = -\frac{Mg}{RT} dz,\ \ g = \frac{Gm_E}{r^2};$$

$$\frac{dP}{P} = -\frac{MGm_E}{RTr^2} dx;\ z = r - r_E;\ (r_E \text{ — радиус Земли}).$$

$$\frac{dP}{P} = -\frac{MGm_E}{RT(z+r_E)^2} dz;\ \ln\frac{P}{P_0} = \frac{MGm_E}{RT}\left(\frac{1}{z+r_E} - \frac{1}{r_E}\right).$$

$$\frac{dP}{P} = -\frac{Mg}{RT} dz = -\frac{Mg}{R(T_0-az)} dz.$$

Пусть $T - az = x$. Тогда $dx = -a\,dz$, а $dz = -dx/a$.

$$\int_{x_0}^{x} -\frac{dx}{ax} = -\frac{1}{a}\ln\frac{x}{x_0} = -\frac{1}{a}\ln\frac{T_0}{T_0-az};\ \ln\frac{P}{P_0} = \frac{Mg}{Ra}\ln\frac{T_0-az}{T_0}.$$

<u>Компьютерное решение</u>.

```
(%i1) 'diff(P,z)=-P*M*g/(R*T);
```

$$\text{(\%o1)}\quad \frac{d}{dz} P = -\frac{g\,M\,P}{R\,T}$$

```
(%i2) ode2(%,P,z); ic1(%,z=0,P=P0);
```

$$\text{(\%o2)}\quad P = \%c\ \%e^{-\frac{g\,z\,M}{R\,T}}$$

$$\text{(\%o3)}\quad P = P0\ \%e^{-\frac{g\,z\,M}{R\,T}}$$

```
(%i4) 'diff(P,z)=-P*M*g/(R*(T0-a*z));
```

$$\text{(\%o4)}\quad \frac{d}{dz} P = -\frac{g\,M\,P}{R\,(T0 - a\,z)}$$

```
(%i5) ode2(%,P,z); ic1(%,z=0,P=P0);
```

$$\text{(\%o5)}\quad P = \%c\ \%e^{\frac{g\,M\,\log(R\,T0 - a\,z\,R)}{a\,R}}$$

$$\text{(\%o6)}\quad P = P0\ \%e^{\frac{g\,M\,\log(R\,T0 - a\,z\,R)}{a\,R} - \frac{g\,M\,\log(R\,T0)}{a\,R}}$$

```
(%i7) radcan(%);
```

$$\text{(\%o7)}\quad P = \frac{P0\,(T0 - a\,z)^{\frac{g\,M}{a\,R}}}{T0^{\frac{g\,M}{a\,R}}}$$

Задача 6 (I-2-18-19 из [2]). Температура Бойля T_B для газа — это температура, при которой $[\partial(PV)/\partial P]_T = 0$, если $P = 0$. Вычислите её для газа Ван-дер-Ваальса:

$$\left(P + \frac{a}{V^2}\right)(V - b) = RT.$$

Традиционное решение

$$P = \frac{RT}{V-b} - \frac{a}{V^2},$$

$$PV = RT\frac{V}{V-b} - \frac{a}{V},$$

$$\left(\frac{\partial(PV)}{\partial P}\right)_T = \left(\frac{\partial(PV)}{\partial V}\right)_T \left(\frac{\partial V}{\partial P}\right)_T = 0, \text{ причём } \left(\frac{\partial V}{\partial P}\right)_T \neq 0.$$

$$\left(\frac{\partial(PV)}{\partial V}\right)_T = \frac{RT}{V-b} - \frac{RTV}{(V-b)^2} + \frac{a}{V^2} = 0.$$

$$RT = \frac{a}{b}\left(1 - \frac{b}{V}\right)^2.$$

В точке Бойля $T = T_B$, $P = 0$ и $V \to \infty$, поэтому $RT_B = a/b$ и $T_B = a/(Rb)$.

Компьютерное решение.

```
(%i1) (p+a/V^2)*(V-b)=R*T$
(%i2) solve(%,p);
(%o2) [p=(R T V^2 - a V + a b)/(V^3 - b V^2)]
(%i3) rhs(first(%))*V;
(%o3) V (R T V^2 - a V + a b)/(V^3 - b V^2)
(%i4) diff(%,V);
(%o4) (R T V^2 - a V + a b)/(V^3 - b V^2) + V (2 R T V - a)/(V^3 - b V^2) - V (3 V^2 - 2 b V)(R T V^2 - a V + a b)/(V^3 - b V^2)^2
(%i5) limit(%, V, inf);
(%o5) 0
```

Задача 7 (I-3-9 из [2]). Для некоторого газа коэффициент термического расширения определяется так:

$$\alpha = \frac{1}{V}\left(\frac{\partial V}{\partial T}\right)_p = k_1 \frac{C_p}{C_v} T^{\left(\frac{C_p}{C_v}\right)-1},$$

а коэффициент изотермического сжатия —

$$\beta = -\frac{1}{V}\left(\frac{\partial V}{\partial P}\right)_T = \frac{k_2}{P}.$$

причём C_p, C_v, k_1 и k_2 — постоянные. Какому уравнению состояния подчиняется этот газ?

Традиционное решение: $dV = \left(\frac{\partial V}{\partial p}\right)_T dP + \left(\frac{\partial V}{\partial T}\right)_p dT.$

Из выражения для α:

$$\left(\frac{\partial V}{\partial T}\right)_p = \frac{k_1 V C_p T^{\left(\frac{C_p}{C_v}\right)-1}}{C_v}.$$

Из выражения для β:

$$\left(\frac{\partial V}{\partial p}\right)_T = -\frac{k_2 V}{P};$$

$$\frac{dV}{V} = -k_2\frac{dP}{P} + k_1\frac{C_p}{C_v}T^{\left(\frac{C_p}{C_v}\right)-1}dT;$$

$$\ln V = -k_2\ln P + k_1\frac{C_p}{C_v}\cdot\frac{C_v}{C_p}T^{\left(\frac{C_p}{C_v}\right)} + const = -k_2\ln P + k_1 T^{\left(\frac{C_p}{C_v}\right)} + const.$$

Таким образом, уравнение состояния будет иметь вид:

$$P^{k_2}V = k\exp\left(k_1 T^{C_p/C_v}\right).$$

Компьютерное решение.

```
(%i1) alpha:k1*(Cp/Cv)*T^((Cp/Cv)-1)$ beta:k2/P$
(%i3) depends(V,[P,T])$ assume(notequal(Cp/Cv,0))$
(%i5) 'diff(V)=alpha*V*del(T)-beta*V*del(P);
```

(%o5) $\mathrm{del}(V)=\dfrac{Cp\,k1\,T^{Cp/Cv-1}\,V\,\mathrm{del}(T)}{Cv}-\dfrac{k2\,V\,\mathrm{del}(P)}{P}$

```
(%i6) 'diff(V,T)=alpha*V; 'diff(V,P)=-beta*V;
```

(%o6) $\dfrac{d}{d\,T}\,V=\dfrac{Cp\,k1\,T^{Cp/Cv-1}\,V}{Cv}$

(%o7) $\dfrac{d}{d\,P}\,V=-\dfrac{k2\,V}{P}$

```
(%i8) ode2(%th(2),V,T); ode2(%th(2),V,P);
```

(%o8) $V=\%c\,\%e^{k1\,T^{Cp/Cv}}$

(%o9) $V=\%c\,\%e^{-k2\log(P)}$

```
(%i10) rhs(%th(2))*rhs(%);
```

(%o10) $\%c^2\,\%e^{k1\,T^{Cp/Cv}-k2\log(P)}$

```
(%i11) V=subst([%c=1],%),radcan;
```

(%o11) $V=\dfrac{\%e^{k1\,T^{Cp/Cv}}}{P^{k2}}$

Задача 8 (III-3-17 из [2]). Покажите, что для газа, подчиняющегося уравнению состояния $PV = RT - BP + APT$, где V — мольный объём, A и B — характеристические постоянные газа, выполняется соотношение

$$C_p - C_V = R\left(1+\frac{AP}{R}\right)^2.$$

Традиционное решение.

$$C_p - C_V = T\left(\frac{\partial P}{\partial T}\right)_V \left(\frac{\partial V}{\partial T}\right)_P;$$

$$P = \frac{RT}{V-AT+B};$$

$$\left(\frac{\partial P}{\partial T}\right)_V = \frac{R}{V-AT+B} + \frac{ART}{(V-AT+B)^2} =$$

$$= \frac{RV-ART+RB+ART}{(V-AT+B)^2} = \frac{R(V+B)}{(V-AT+B)^2};$$

$$V = \frac{RT}{P} + AT - B;$$

$$\left(\frac{\partial V}{\partial T}\right)_P = \frac{R}{P} + A;$$

$$C_p - C_V = \frac{RT(V+B)(\frac{R}{P}+A)}{(V-AT+B)^2} = \frac{RT^2\left(\frac{R}{P}-A\right)^2}{\left(RT/_P\right)^2} =$$

$$= \frac{P^2}{R}\left(\frac{R}{P}+A\right)^2 = R\left(1+\frac{AP}{R}\right)^2.$$

Компьютерное решение.

```
(%i1) p*V-R*T+B*p-A*p*T$
(%i2) T*diff(rhs(first(solve(%,p))),T)*diff(rhs(first(solve(%,V))),T);
```

$$\text{(\%o2)}\quad \frac{(R+p\,A)\,T\left(\frac{R}{V-A\,T+B}+\frac{A\,R\,T}{(V-A\,T+B)^2}\right)}{p}$$

```
(%i3) solve(%th(2),V);
```

$$\text{(\%o3)}\quad [\,V=\frac{(R+p\,A)\,T-p\,B}{p}]$$

```
(%i4) subst(%,%th(2)),factor;
```

$$\text{(\%o4)}\quad \frac{(R+p\,A)^2}{R}$$

Задача 9 (III-2-8 из [2]). Для некоторого процесса $\Delta G^0{}_{298} = B$, $\Delta H_0 = D + aT + bT^2 + cT^3$. Выведите выражение для ΔG^0 как функции температуры.

Традиционное решение.

$$\left[\frac{\partial(\Delta G^0/T)}{\partial T}\right]_P = -\frac{\Delta H^0}{T^2} = -\frac{D}{T^2} - \frac{a}{T} - b - cT;$$

$$\int_{\frac{\Delta G^0}{T_1}}^{\frac{\Delta G^0}{T_2}} d\left(\frac{\Delta G^0}{T}\right) = -\int_{298}^{T}\frac{DdT}{T^2} - \int_{298}^{T}\frac{adT}{T} - \int_{298}^{T} bdT - \int_{298}^{T} cTdT;$$

$$\frac{\Delta G^0}{T} = \frac{\Delta G_{298}}{298} + D\left(\frac{1}{T}-\frac{1}{298}\right) - a\ln\frac{T}{298} - b(T-298) - \frac{c}{2}(T^2-298^2);$$

$$\Delta G^0 = T\left[\frac{B}{298} - \frac{D}{298} + 2.303a\lg 298 + 298b + \frac{c}{2}298^2\right] + D - 2.3aR\lg T -$$

$$bT2 - cT32.$$

Компьютерное решение.

```
(%i1) H:D+a*T+b*T^2+c*T^3$
(%i2) depends(G,T)$ H=G-T*diff(G,T);
```

(%o3) $c\,T^3+b\,T^2+a\,T+D=G-\left(\frac{d}{d\,T}\,G\right)T$

```
(%i4) ode2(%,G,T); ic1(%,T=298,G=G298);
```

(%o4) $G=T\left(-a\log(T)-\frac{c\,T^2+2\,b\,T}{2}+\frac{D}{T}+\%c\right)$

(%o5) $G=-\frac{298\,a\,T\log(T)+149\,c\,T^3+298\,b\,T^2+(-G298+D-13231796\,c-88804\,b-298\log(298)\,a)\,T-298}{298}$

```
(%i6) expand(%);
```

(%o6) $G=-a\,T\log(T)-\frac{c\,T^3}{2}-b\,T^2+\frac{G298\,T}{298}-\frac{D\,T}{298}+44402\,c\,T+298\,b\,T+\log(298)\,a\,T+D$

Задача 10 (III-2-26 из [2]). Газ подчиняется уравнению состояния $PV = RT + aP^{1/2} + bP + cP^{3/2}$. Выведите уравнение для коэффициента летучести этого газа α при постоянной температуре T.

Традиционное решение.

$$d\ln\gamma = \frac{\alpha}{RT}dP;\ \ \alpha = V - \frac{RT}{P};\ \ V = \frac{RT}{P} + aP^{-1/2} + b + cP^{1/2};$$

$$\alpha = \frac{RT}{P} + aP^{-1/2} + b + cP^{1/2} - \frac{RT}{P};$$

$$\ln\gamma = \int_0^{\ln\gamma} d\ln\gamma = \frac{1}{RT}\int_0^P\left(aP^{-1/2} + b + cP^{1/2}\right)dP = \frac{1}{RT}\Big(2aP^{1/2} + bP +$$

$$23cP3/2.$$

Компьютерное решение.

```
(%i1) assume(P>0)$
(%i2) p*V=R*T+a*p^(1/2)+b*p+c*p^(3/2)$
(%i3) solve(%,V);
```

(%o3) $\left[V=\frac{R\,T+c\,p^{3/2}+b\,p+a\sqrt{p}}{p}\right]$

```
(%i4) (rhs(first(%))-R*T/p)/(R*T),ratsimp;
```

(%o4) $\frac{\sqrt{p}\,(c\,p+a)+b\,p}{p\,R\,T}$

```
(%i5) integrate(%,p,0,P),expand;
```

(%o5) $\frac{2\,c\,P^{3/2}}{3\,R\,T}+\frac{b\,P}{R\,T}+\frac{2\,a\sqrt{P}}{R\,T}$

Задача 11 (III-2-27 из [2]). Газ подчиняется уравнению состояния

$$\bar{V} = \frac{RT}{P}(1 + AP + BP^2),$$

где A и B — функции только температуры. Определите коэффициент летучести $\gamma = f/P$ газа через величины P, A, B и универсальные постоянные.

Традиционное решение.

$$\frac{\bar{V}}{RT} = \frac{1}{P} + A + BP,$$

$$\ln\gamma = \int_0^P \left(\frac{\bar{V}}{RT} - \frac{1}{P}\right) dP = \int_0^P (A + BP) dP = AP + \frac{1}{2} BP^2,$$

$$\gamma = \exp\left(AP + \frac{1}{2} BP^2\right).$$

Компьютерное решение.

```
(%i1) V:R*T*(1+A*p+B*p^2)/p$
(%i2) V/(R*T)-1/p,expand;
(%o2) p B+A
(%i3) integrate(%,p,0,P);
```

$$\text{(\%o3)} \quad \frac{B\,P^2+2\,A\,P}{2}$$

```
(%i4) exp(%);
```

$$\text{(\%o4)} \quad \%e^{\frac{B\,P^2+2\,A\,P}{2}}$$

2. Элементы теории вероятностей

Задача 1 (1 к главе II из [3]). Материальная точка колеблется по закону $x = \sin \omega t$. Найти вероятность того, что при однократном измерении её положение находится в интервале $[x, x + \mathrm{d}x]$.

Традиционное решение. Поскольку материальная точка совершает периодическое движение, за полное время наблюдения можно взять период колебаний T. Если $\mathrm{d}t$ — время, в течение которого точка находится в интервале $[x, x + \mathrm{d}x]$, то искомая вероятность

$$dw = \frac{2dt}{T}.$$

Множитель 2 введён потому, что за период T точка побывает в указанном интервале дважды.

Выражая время $\mathrm{d}t$ через $\mathrm{d}x$, имеем:

$$dt = \frac{1}{\omega}(\arcsin x)'dx = \frac{dx}{\omega\sqrt{1-x^2}}.$$

Следовательно,

$$dw = \frac{1}{\pi}\cdot\frac{dx}{\sqrt{1-x^2}}.$$

Компьютерное решение.

```
(%i1) x=sin(omega*t)$
(%i2) solve(%,t);
solve: using arc-trig functions to get a solution.
 Some solutions will be lost.
```

$$\text{(\%o2)}\quad [t=\frac{\operatorname{asin}(x)}{\omega}]$$

```
(%i3) diff(rhs(first(%)),x);
```

$$\text{(\%o3)}\quad \frac{1}{\omega\sqrt{1-x^2}}$$

```
(%i4) %*2/T;
```

$$\text{(\%o4)}\quad \frac{2}{\omega\sqrt{1-x^2}\,T}$$

```
(%i5) subst([omega=%pi,T=2],%);
```

$$\text{(\%o5)}\quad \frac{1}{\pi\sqrt{1-x^2}}$$

Обратите внимание на замену в последней строке: так как, ввиду разнесённости ω и T выше (Maxima автоматически упорядочивает переменные в результирующем выражении по-своему), мы не можем сделать одну замену $\omega T \to 2\pi$ (полный период), то делаем её в два приёма.

Задача 2 (3 к главе II из [3]). Система характеризуется распределением вероятностей

$$dw \sim xydxdy,$$

где переменные x и y лежат в интервале $0 \le x \le a$, $0 \le y \le b$. Нормируйте распределение вероятностей.

Традиционный вариант. Значение коэффициента пропорциональности определим из условия нормировки

$$w = \int_0^a \int_0^b c\, xy\, dx\, dy = c\frac{a^2}{2}\frac{b^2}{2} = 1, \quad c = \frac{4}{a^2b^2}, \text{ откуда}$$

$$dw = \frac{4}{a^2b^2}xydxdy.$$

Компьютерное решение.

```
(%i1) p(x,y):=x*y$
(%i2) integrate(integrate(p(x,y),x,0,a),y,0,b);
(%o2) a^2 b^2/4
(%i3) p(x,y)/%;
(%o3) 4 x y/(a^2 b^2)
```

Задача 3 (4 к главе II из [3]). Найти вероятность того, что вышеуказанная система имеет данное значение величины x при любом значении величины y.

Традиционное решение:

$$dw = \frac{4}{a^2b^2}xdx \int_0^b y\, dy = \frac{2}{a^2}xdx.$$

Компьютерное решение (в дополнение к предыдущему листингу).

```
(%i4) integrate(%,y,0,b);
(%o4) 2 x/a^2
```

Задача 4 (5 к главе II из [3]). Найти среднее значение величины x, её среднее квадратичное значение, среднюю квадратичную флуктуацию и относительную флуктуацию, если

$$dw \sim e^{-\alpha x}dx.$$

Традиционное решение:

$$dw = \alpha e^{-\alpha x}dx,$$

$$\bar{x} = \frac{1}{\alpha},$$

$$\overline{x^2} = \frac{2}{\alpha^2},$$

$$\overline{(\Delta x)^2} = \frac{2}{\alpha^2} - \frac{1}{\alpha^2} = \frac{1}{\alpha^2},$$

$$\delta x = \frac{\sqrt{\overline{(\Delta x)^2}}}{\bar{x}} = 1.$$

Компьютерное решение.

```
(%i1) assume(alpha>0)$ temp:exp(-alpha*x)$
(%i3) p(x):=temp/integrate(temp,x,0,inf)$ p(x);
(%o4) α %e^(-α x)
(%i5) integrate(x*p(x),x,0,inf);
      1
(%o5) —
      α
(%i6) integrate(x^2*p(x),x,0,inf)-%^2;
       1
(%o6) ——
       2
      α
(%i7) sqrt(%)/%th(2);
(%o7) 1
```

Задача 5 (6 к главе II из [3]). Найти распределения вероятностей для той же системы, если её характеризовать величиной y, связанной с x соотношением $y^2 = x$.

Традиционно: $dw = 2\alpha e^{-\alpha y^2} y dy$.

Компьютерное решение (в дополнение к предыдущему листингу).

```
(%i8) subst([x=y^2],p(x))*diff(y^2,y);
(%o8) 2 α y %e^(-α y^2)
```

Другой вариант компьютерного решения для задач 4 и 5.

```
(%i1) assume(alpha>0)$
(%i2) C*integrate(exp(-alpha*x),x,0,inf)=1;
      C
(%o2) —=1
      α
(%i3) solve(%,C);
(%o3) [C=α]
(%i4) rhs(first(%))*exp(-alpha*x);
(%o4) α %e^(-α x)
(%i5) subst([x=y^2,dx=2*y*dy],%*dx);
(%o5) 2 α dy y %e^(-α y^2)
```

Задача 6 (2.22 из [1]). Найти $\bar{x}$, $\overline{x^2}$, и $\overline{(\Delta x)^2}$ для нормированного гауссова распределения $f(x) = \sqrt{\frac{\alpha}{\pi}} e^{-\alpha x^2}$ $(-\infty < x < +\infty)$.

Традиционно: Среднее значение величины x

$$\bar{x} = \int_{-\infty}^{\infty} \sqrt{\frac{\alpha}{\pi}} x e^{-\alpha x^2} dx = 0,$$

$$\overline{x^2} = \int_{-\infty}^{\infty} \sqrt{\frac{\alpha}{\pi}} x^2 e^{-\alpha x^2} dx = \sqrt{\frac{\alpha}{\pi}} \int_{-\infty}^{\infty} x^2 e^{-\alpha x^2} dx = \sqrt{\frac{\alpha}{\pi}} \frac{1}{2} \sqrt{\frac{\pi}{\alpha^3}} = \frac{1}{2\alpha},$$

$$\overline{(\Delta x)^2} = \overline{x^2} - (\bar{x})^2 = \frac{1}{2\alpha}.$$

Компьютерное решение.

```
(%i1) assume(alpha>0)$
(%i2) f(x):=sqrt(alpha/%pi)*exp(-alpha*x^2)$
(%i3) create_list(integrate(x^n*f(x),x,-inf,inf),n,[0,1,2]);
(%o3) [1,0,1/(2 α)]
```

Задача 7 (2.23 из [1]). Плотность распределения вероятности случайных амплитуд боковой качки корабля имеет вид (закон Релея)

$$f(x) = \frac{x}{a^2} e^{-x^2/2a^2} \quad (x \geq 0),$$

где a — максимальное отклонение корабля. Найти среднее отклонение корабля во время качки.

Традиционно. Среднее отклонение корабля по вертикали

$$\bar{x} = \frac{1}{a^2} \int_0^{\infty} x^2 \, e^{-x^2/2a^2} dx.$$

Произведя замену переменных ${}^{x}/_{a\sqrt{2}} = t$, получим

$$\bar{x} = 2\sqrt{2}a \int_0^{\infty} t^2 e^{-t^2} dt = 2\sqrt{2} \frac{\sqrt{\pi}}{4} a = a\sqrt{\frac{\pi}{2}}.$$

Компьютерное решение.

```
(%i1) assume(a>0)$ temp:x*exp(-x^2/(2*a^2))$
(%i3) f(x):=temp/integrate(temp,x,0,inf)$ f(x);
(%o4) x %e^(-x^2/(2 a^2)) / a^2
(%i5) integrate(x*f(x),x,0,inf);
(%o5) sqrt(π) a / sqrt(2)
```

Задача 8 (задача складного метра из [4]). Пусть складной метр состоит из N шарнирно соединённых сегментов длиной l. Начало первого сегмента закреплено в точке $x = 0$. Любой сегмент с равной вероятностью (1/2) может располагаться слева направо и справа налево. Найти функцию распределения вероятности $dW_{\{x,x+dx\}} = f(x)dx$, т.е. вероятность того, что расстояние между концами складного метра после размещения всех сегментов находится в пределах $\{x, x + dx\}$.

Традиционное решение дано в учебнике [4], раздел 4.3, с. 114–120.

Компьютерное решение.

```
(%i1) load (stirling)$ assume(N>0,l>0)$
(%i3) gamma((N+x/l)/2+1)*gamma((N-x/l)/2+1)$
(%i4) stirling(%,1);
```

$$(\%o4)\quad 2\,\pi\left(\frac{N-\frac{x}{l}}{2}+1\right)^{\frac{N-\frac{x}{l}}{2}+\frac{1}{2}}\left(\frac{N+\frac{x}{l}}{2}+1\right)^{\frac{N+\frac{x}{l}}{2}+\frac{1}{2}}\%e^{-\frac{N+\frac{x}{l}}{2}-\frac{N-\frac{x}{l}}{2}-2}$$

```
(%i5) log(%), logexpand=super;
```

$$(\%o5)\quad \left(\frac{N+\frac{x}{l}}{2}+\frac{1}{2}\right)\log\left(\frac{N+\frac{x}{l}}{2}+1\right)+\left(\frac{N-\frac{x}{l}}{2}+\frac{1}{2}\right)\log\left(\frac{N-\frac{x}{l}}{2}+1\right)-\frac{N+\frac{x}{l}}{2}-\frac{N-\frac{x}{l}}{2}+\log(\pi)+\log(2)-2$$

```
(%i6) pickapart(%,1);
```

$$(\%t6)\quad \left(\frac{N+\frac{x}{l}}{2}+\frac{1}{2}\right)\log\left(\frac{N+\frac{x}{l}}{2}+1\right)$$

$$(\%t7)\quad \left(\frac{N-\frac{x}{l}}{2}+\frac{1}{2}\right)\log\left(\frac{N-\frac{x}{l}}{2}+1\right)$$

$$(\%t8)\quad -\frac{N+\frac{x}{l}}{2}$$

$$(\%t9)\quad -\frac{N-\frac{x}{l}}{2}$$

$$(\%t10)\quad \log(\pi)$$

$$(\%t11)\quad \log(2)$$

$$(\%o11)\quad \%t9+\%t8+\%t7+\%t6+\%t11+\%t10-2$$

```
(%i12) %t6+%t7;
```

$$(\%o12)\quad \left(\frac{N+\frac{x}{l}}{2}+\frac{1}{2}\right)\log\left(\frac{N+\frac{x}{l}}{2}+1\right)+\left(\frac{N-\frac{x}{l}}{2}+\frac{1}{2}\right)\log\left(\frac{N-\frac{x}{l}}{2}+1\right)$$

Поскольку N — очень большое число, величины типа ½ или 1, складывающиеся/вычитающиеся с $N/2$, пренебрежимо малы. Они исчезли бы при переходе к пределу $N \to \infty$, но пока оставим N большим, хотя и конечным. Более рационально вручную скопировать последний результат через буфер в новую строку и убрать названные элементы, после чего продолжить компьютерные преобразования.

(%i13) ((N+x/l)/2)*log((N+x/l)/2)+((N-x/l)/2)*log((N-x/l)/2);

(%o13) $$\frac{\left(N+\frac{x}{l}\right)\log\left(\frac{N+\frac{x}{l}}{2}\right)}{2}+\frac{\left(N-\frac{x}{l}\right)\log\left(\frac{N-\frac{x}{l}}{2}\right)}{2}$$

(%i14) taylor(%,x,0,3);

(%o14)/T/ $$(-\log(2)+\log(N))\,N+\frac{x^2}{2\,l^2\,N}+...$$

(%i15) pickapart(%,1);

(%t15) $(-\log(2)+\log(N))\,N$

(%t16) $$\frac{x^2}{2\,l^2\,N}$$

(%o16) $\%t15+\%t16+...$

(%i17) exp(-%t16);

(%o17) $$\%e^{-\frac{x^2}{2\,l^2\,N}}$$

(%i18) integrate(%,x,-inf,inf);

(%o18) $\sqrt{2}\,\sqrt{\pi}\,l\,\sqrt{N}$

(%i19) %th(2)/%;

(%o19) $$\frac{\%e^{-\frac{x^2}{2\,l^2\,N}}}{\sqrt{2}\,\sqrt{\pi}\,l\,\sqrt{N}}$$

3. Элементы классической механики

Задача математического маятника (хорошо известна из общего курса физики [5]).

А) Решение непосредственно по методу Лагранжа.

```
(%i1) depends(phi,t)$
(%i2) Ek:m/2*l^2*(diff(phi,t))^2$
(%i3) U:-m*g*l*cos(phi)$
(%i4) L:Ek-U;
```

(%o4) $$\frac{l^2\, m \left(\frac{d}{d\,t}\phi\right)^2}{2}+g\, l\, m\cos(\phi)$$

```
(%i5) diff(diff(L,diff(phi,t)),t)-diff(L,phi),factor;
```

(%o5) $$l\, m\left(l\left(\frac{d^2}{d\,t^2}\phi\right)+g\sin(\phi)\right)$$

```
(%i6) ode2(%,phi,t);
```

(%o6) $$\left[-\frac{\displaystyle\int \frac{1}{\sqrt{\frac{g\cos(\phi)-\%k1\, g}{l}}}\,d\phi}{\sqrt{2}}=t+\%k2,\ \frac{\displaystyle\int \frac{1}{\sqrt{\frac{g\cos(\phi)-\%k1\, g}{l}}}\,d\phi}{\sqrt{2}}=t+\%k2\right]$$

Последние интегралы не берутся аналитически, поэтому решение приходится на этом этапе закончить (решение может быть только численным).

Б) Аналогичное решение с приведением времени.

Поскольку размерность времени — с, а частоты ω — $с^{-1}$, их перемножением можно получить безразмерную величину, пропорциональную времени — приведённое время τ. Это упрощает выражения и преобразования (хотя интегралы по-прежнему не берутся).

```
(%i7) pickapart(%th(2),1);
```

$$(\%t7)\quad l\left(\frac{d^2}{d\,t^2}\phi\right)+g\sin(\phi)$$

$$(\%o7)\quad \%t7\;l\;m$$

```
(%i8) assume(omega>0)$ omega=sqrt(g/l)$ solve(%,l);
```

$$(\%o10)\quad [\,l=\frac{g}{\omega^2}\,]$$

```
(%i11) subst(%,%t7),factor;
```

$$(\%o11)\quad \frac{g\left(\frac{d^2}{d\,t^2}\phi+\omega^2\sin(\phi)\right)}{\omega^2}$$

```
(%i12) tau=omega*t$ solve(%,t);
```

$$(\%o13)\quad [\,t=\frac{\tau}{\omega}\,]$$

```
(%i14) subst(%,%th(3));
```

$$(\%o14)\quad \frac{g\left(\frac{d^2}{d\,\frac{\tau^2}{\omega^2}}\phi+\omega^2\sin(\phi)\right)}{\omega^2}$$

```
(%i15) pickapart(%,2);
```

$$(\%t15)\quad \frac{d^2}{d\,\frac{\tau^2}{\omega^2}}\phi+\omega^2\sin(\phi)$$

$$(\%o15)\quad \frac{\%t15\;g}{\omega^2}$$

```
(%i16) depends(phi,tau)$ diff(phi,tau,2)+sin(phi);
```

$$(\%o17)\quad \frac{d^2}{d\,\tau^2}\phi+\sin(\phi)$$

```
(%i18) ode2(%,phi,tau);
```

$$(\%o18)\quad [\,-\frac{\int\frac{1}{\sqrt{\cos(\phi)-\%k1}}d\phi}{\sqrt{2}}=\tau+\%k2,\;\frac{\int\frac{1}{\sqrt{\cos(\phi)-\%k1}}d\phi}{\sqrt{2}}=\tau+\%k2\,]$$

Попробуем теперь заменить обобщённую координату: пусть q есть кубичный корень из угла отклонения маятника: $\varphi = q^3$. Снова решаем по методу Лагранжа.

```
(%i1) depends(q,t)$
(%i2) Ek:m/2*l^2*(diff(q^3,t))^2$
(%i3) U:-m*g*l*cos(q^3)$
(%i4) L:Ek-U;
```

$$(\%o4)\quad g\, l\, m \cos(q^3)+\frac{9\, l^2\, m\, q^4 \left(\frac{d}{d\,t} q\right)^2}{2}$$

```
(%i5) diff(diff(L,diff(q,t)),t)-diff(L,q),factor;
```

$$(\%o5)\quad 3\, l\, m\, q^2 \left(g \sin(q^3)+3\, l\, q^2 \left(\frac{d^2}{d\,t^2} q\right)+6\, l\, q \left(\frac{d}{d\,t} q\right)^2\right)$$

```
(%i6) pickpart(%,1);
```

$$(\%t6)\quad q^2$$

$$(\%t7)\quad g \sin(q^3)+3\, l\, q^2 \left(\frac{d^2}{d\,t^2} q\right)+6\, l\, q \left(\frac{d}{d\,t} q\right)^2$$

$$(\%o7)\quad 3\ \%t6\ \%t7\ l\, m$$

```
(%i8) ode2(%t7,q,t);
```

$$(\%o8)\quad \left[-\frac{3\displaystyle\int \frac{q^2}{\sqrt{\frac{g \cos(q^3)+3\ \%k1}{l}}}\,dq}{\sqrt{2}}=t+\%k2,\ \frac{3\displaystyle\int \frac{q^2}{\sqrt{\frac{g \cos(q^3)+3\ \%k1}{l}}}\,dq}{\sqrt{2}}=t+\%k2\right]$$

Теперь за обобщённую координату примем φ – отклонение маятника по горизонтали: φ = *x*. Снова решаем по методу Лагранжа.

```
(%i1) depends(x,t)$
(%i2) Ek:m/2*l^2/(l^2-x^2)*(diff(x,t))^2$
(%i3) U:-m*g*sqrt(l^2-x^2)$
(%i4) L:Ek-U;
```

$$(\%o4)\quad \frac{l^2\, m\left(\frac{d}{d\,t}\,x\right)^2}{2\,(l^2-x^2)}+g\,m\sqrt{l^2-x^2}$$

```
(%i5) diff(diff(L,diff(x,t)),t)-diff(L,x),factor;
```

$$(\%o5)\quad -\frac{m\left(l^2\,x^2\sqrt{l^2-x^2}\left(\frac{d^2}{d\,t^2}\,x\right)-l^4\sqrt{l^2-x^2}\left(\frac{d^2}{d\,t^2}\,x\right)-l^2\,x\sqrt{l^2-x^2}\left(\frac{d}{d\,t}\,x\right)^2-g\,x^5+2\,g\,l^2\,x^3-g\,l^4\,x\right)}{(x-l)^2\,(x+l)^2\sqrt{l^2-x^2}}$$

```
(%i6) pickapart(%,3);
```

$$(\%t6)\quad l^2\,x^2\sqrt{l^2-x^2}\left(\frac{d^2}{d\,t^2}\,x\right)-l^4\sqrt{l^2-x^2}\left(\frac{d^2}{d\,t^2}\,x\right)-l^2\,x\sqrt{l^2-x^2}\left(\frac{d}{d\,t}\,x\right)^2-g\,x^5+2\,g\,l^2\,x^3-g\,l^4\,x$$

$$(\%t7)\quad (x-l)^2$$

$$(\%t8)\quad (x+l)^2$$

$$(\%t9)\quad \sqrt{l^2-x^2}$$

$$(\%o9)\quad -\frac{\%t6\;m}{\%t7\;\%t8\;\%t9}$$

```
(%i10) ode2(%t6,x,t);
```

$$(\%o10)\quad \left[-\frac{l\displaystyle\int\frac{1}{(x^2-l^2)\sqrt{\frac{g\sqrt{x^2-l^2}+\%k1}{\sqrt{l^2-x^2}\,\sqrt{x^2-l^2}}}}\,d\,x}{\sqrt{2}}=t+\%k2,\;\frac{l\displaystyle\int\frac{1}{(x^2-l^2)\sqrt{\frac{g\sqrt{x^2-l^2}+\%k1}{\sqrt{l^2-x^2}\,\sqrt{x^2-l^2}}}}\,d\,x}{\sqrt{2}}=t+\%k2\right]$$

Задача о маятнике с равномерно движущейся точкой подвеса [5].

```
(%i1) depends([phi,x,y],t)$
(%i2) x=l*sin(phi)+v*t$ y=l*cos(phi)$
(%i4) diff(%th(2),t); diff(%th(2),t);
```

(%o4) $\frac{d}{d\,t}x=v+l\cos(\phi)\left(\frac{d}{d\,t}\phi\right)$

(%o5) $\frac{d}{d\,t}y=-l\sin(\phi)\left(\frac{d}{d\,t}\phi\right)$

```
(%i6) m/2*(diff(x,t)^2+diff(y,t)^2)$
(%i7) Ek:subst([%th(2),%th(3)],%);
```

(%o7) $\frac{m\left(\left(v+l\cos(\phi)\left(\frac{d}{d\,t}\phi\right)\right)^2+l^2\sin(\phi)^2\left(\frac{d}{d\,t}\phi\right)^2\right)}{2}$

```
(%i8) U:-m*g*l*cos(phi)$ L:Ek-U;
```

(%o9) $\frac{m\left(\left(v+l\cos(\phi)\left(\frac{d}{d\,t}\phi\right)\right)^2+l^2\sin(\phi)^2\left(\frac{d}{d\,t}\phi\right)^2\right)}{2}+g\,l\,m\cos(\phi)$

```
(%i10) diff(diff(L,diff(phi,t)),t)-diff(L,phi),expand;
```

(%o10) $l^2\,m\sin(\phi)^2\left(\frac{d^2}{d\,t^2}\phi\right)+l^2\,m\cos(\phi)^2\left(\frac{d^2}{d\,t^2}\phi\right)+g\,l\,m\sin(\phi)$

```
(%i11) trigreduce(%),factor;
```

(%o11) $l\,m\left(l\left(\frac{d^2}{d\,t^2}\phi\right)+g\sin(\phi)\right)$

```
(%i12) pickapart(%,1);
```

(%t12) $l\left(\frac{d^2}{d\,t^2}\phi\right)+g\sin(\phi)$

(%o12) $\%t12\,l\,m$

```
(%i13) ode2(%t12,phi,t);
```

(%o13) $\left[-\frac{\int\frac{1}{\sqrt{\frac{g\cos(\phi)-\%k1\,g}{l}}}d\phi}{\sqrt{2}}=t+\%k2,\ \frac{\int\frac{1}{\sqrt{\frac{g\cos(\phi)-\%k1\,g}{l}}}d\phi}{\sqrt{2}}=t+\%k2\right]$

Переходим к уравнениям Гамильтона. Повторим рассмотрение математического маятника. Получаем для него уравнения Гамильтона,

```
(%i1) depends([phi,p],t)$ assume(g>0,l>0)$
(%i3) Ek:m/2*l^2*(diff(phi,t))^2$
(%i4) U:-m*g*l*cos(phi)$
(%i5) L:Ek-U;
```

$$(\%o5)\quad \frac{l^2\,m\left(\frac{d}{d\,t}\phi\right)^2}{2}+g\,l\,m\cos(\phi)$$

```
(%i6) p=diff(L,diff(phi,t))$
(%i7) solve(%,diff(phi,t));
```

$$(\%o7)\quad \left[\frac{d}{d\,t}\phi=\frac{p}{l^2\,m}\right]$$

```
(%i8) Ek+U;
```

$$(\%o8)\quad \frac{l^2\,m\left(\frac{d}{d\,t}\phi\right)^2}{2}-g\,l\,m\cos(\phi)$$

```
(%i9) H:subst(%th(2),%);
```

$$(\%o9)\quad \frac{p^2}{2\,l^2\,m}-g\,l\,m\cos(\phi)$$

```
(%i10) diff(phi,t)=diff(H,p);
```

$$(\%o10)\quad \frac{d}{d\,t}\phi=\frac{p}{l^2\,m}$$

```
(%i11) diff(p,t)=-diff(H,phi);
```

$$(\%o11)\quad \frac{d}{d\,t}p=-g\,l\,m\sin(\phi)$$

решаем их

```
(%i12) diff(%th(2),t);
```

$$(\%o12)\quad \frac{d^2}{d\,t^2}\phi=\frac{\frac{d}{d\,t}p}{l^2\,m}$$

```
(%i13) subst([%th(2)],%);
```

$$(\%o13)\quad \frac{d^2}{d\,t^2}\phi=-\frac{g\sin(\phi)}{l}$$

```
(%i14) ode2(%,phi,t);
```

$$(\%o14)\quad \left[-\frac{\sqrt{l}\int\frac{1}{\sqrt{\cos(\phi)-\%k1}}d\phi}{\sqrt{2}\sqrt{g}}=t+\%k2,\ \frac{\sqrt{l}\int\frac{1}{\sqrt{\cos(\phi)-\%k1}}d\phi}{\sqrt{2}\sqrt{g}}=t+\%k2\right]$$

и видим, что аналитическое решение невозможно. Поэтому применяем приближение малых углов отклонения, когда $\sin\varphi\approx\varphi$.

(%i15) subst([sin(phi)=phi],%th(2));

(%o15) $$\frac{d^2}{d\,t^2}\phi=-\frac{g\,\phi}{l}$$

(%i16) ode2(%,phi,t);

(%o16) $$\phi=\%k1\,\sin\left(\frac{\sqrt{g}\,t}{\sqrt{l}}\right)+\%k2\,\cos\left(\frac{\sqrt{g}\,t}{\sqrt{l}}\right)$$

(%i17) ic2(%,t=0,phi=phi0,diff(phi,t)=0);

(%o17) $$\phi=phi0\,\cos\left(\frac{\sqrt{g}\,t}{\sqrt{l}}\right)$$

(%i18) solve(%th(8),p);

(%o18) $$[p=l^2\,m\left(\frac{d}{d\,t}\phi\right)]$$

(%i19) subst([%th(2)],rhs(first(%)));

(%o19) $$l^2\,m\left(\frac{d}{d\,t}\left(phi0\,\cos\left(\frac{\sqrt{g}\,t}{\sqrt{l}}\right)\right)\right)$$

Здесь нужно вмешаться («помочь»). Мышкой выделяем производную внутри скобок, копируем в новую ячейку, снимаем знак апострофа и слева вводим “p=”. Поиск решения продолжается.

(%i20) p=diff((phi0*cos((sqrt(g)*t)/sqrt(l))),t,1);

(%o20) $$p=-\frac{\sqrt{g}\,phi0\,\sin\left(\frac{\sqrt{g}\,t}{\sqrt{l}}\right)}{\sqrt{l}}$$

(%i21) solve(%th(4),t);

solve: using arc-trig functions to get a solution.
 Some solutions will be lost.

(%o21) $$[t=\frac{\sqrt{l}\,\mathrm{acos}\left(\frac{\phi}{phi0}\right)}{\sqrt{g}}]$$

(%i22) subst(%,%th(2));

(%o22) $$p=-\frac{\sqrt{g}\,\sqrt{1-\frac{\phi^2}{phi0^2}}\,phi0}{\sqrt{l}}$$

Таким образом, исключив из уравнений время, получили уравнение фазовой траектории, связывающее обобщённую координату φ и обобщённый импульс *p*.

В завершение проанализируем маятник Капицы: здесь точка подвеса вибрирует в вертикальном направлении [6].

```
(%i1) depends([phi,x,y],t)$
(%i2) x=l*sin(phi)$ y=-l*cos(phi)-a*cos(omega0*t)$
(%i4) diff(%th(2),t); diff(%th(2),t);
```

(%o4) $\frac{d}{d\,t}x = l\cos(\phi)\left(\frac{d}{d\,t}\phi\right)$

(%o5) $\frac{d}{d\,t}y = a\,omega0\,\sin(omega0\,t) + l\sin(\phi)\left(\frac{d}{d\,t}\phi\right)$

```
(%i6) m/2*(diff(x,t)^2+diff(y,t)^2)$
(%i7) Ek:subst([%th(3),%th(2)],%)$
(%i8) U:-m*g*(l*cos(phi)+a*cos(omega0*t))$
(%i9) L:Ek-U;
```

(%o9) $\frac{m\left(\left(a\,omega0\,\sin(omega0\,t)+l\sin(\phi)\left(\frac{d}{d\,t}\phi\right)\right)^2 + l^2\cos(\phi)^2\left(\frac{d}{d\,t}\phi\right)^2\right)}{2} + g\,m\,(a\cos(omega0\,t)+l\cos(\phi))$

```
(%i10) diff(diff(L,diff(phi,t)),t)-diff(L,phi),expand,factor;
```

(%o10) $l\,m\left(a\,omega0^2\sin(\phi)\cos(omega0\,t) + l\sin(\phi)^2\left(\frac{d^2}{d\,t^2}\phi\right) + l\cos(\phi)^2\left(\frac{d^2}{d\,t^2}\phi\right) + g\sin(\phi)\right)$

```
(%i11) pickapart(%,1);
```

(%t11) $a\,omega0^2\sin(\phi)\cos(omega0\,t) + l\sin(\phi)^2\left(\frac{d^2}{d\,t^2}\phi\right) + l\cos(\phi)^2\left(\frac{d^2}{d\,t^2}\phi\right) + g\sin(\phi)$

(%o11) $\%t11\ l\ m$

Видим нераскрытое выражение $\sin^2+\cos^2$, нужно вмешаться. Копируем эту часть выражения вручную, и всё налаживается.

```
(%i12) l*sin(phi)^2*('diff(phi,t,2))+l*cos(phi)^2*('diff(phi,t,2)),trigreduce;
```

(%o12) $l\left(\frac{d^2}{d\,t^2}\phi\right)$

4. Фазовое пространство

Расчёт фазового объёма, в котором находятся изображающие точки частиц идеального газа с энергией, не больше заданной E.

Объём энергетического слоя, отвечающего состояниям частицы с энергией в диапазоне $[E, E+\mathrm{d}E]$, получаем интегрированием элементарного фазового объёма по всем состояниям в этом слое. Интегрирование элементарного фазового объёма по подпространству координат даёт объём V, для интегрирования по подпространству импульсов воспользуемся связью между импульсом и кинетической энергией:

$$E = \frac{p^2}{2m},$$

где p — импульс частицы, E — её кинетическая энергия, m — масса. Следовательно, в подпространстве импульсов состояния с энергией E отображаются точками, лежащими на сфере соответствующего радиуса. Объём шарового слоя толщиной $\mathrm{d}p$ равен $4\pi p^2\mathrm{d}p$. Переводим импульс в энергию и интегрируем полученное выражение от 0 до E.

```
(%i1) assume(E>0)$ E=p^2/(2*m)$
(%i3) dgdp:V*4*%pi*p^2$
(%i4) solve(%th(2),p);
```

(%o4) $[p=-\sqrt{2}\sqrt{m}\sqrt{E}, p=\sqrt{2}\sqrt{m}\sqrt{E}]$

```
(%i5) second(%);
```

(%o5) $p=\sqrt{2}\sqrt{m}\sqrt{E}$

```
(%i6) diff(rhs(%),E);
```

(%o6) $\dfrac{\sqrt{m}}{\sqrt{2}\sqrt{E}}$

```
(%i7) subst([%th(2)],%th(4))*%,radcan;
```

(%o7) $2^{5/2}\pi m^{3/2}\sqrt{E}\,V$

```
(%i8) integrate(%,E,0,E);
```

(%o8) $\dfrac{2^{7/2}\pi m^{3/2}E^{3/2}V}{3}$

Обратите внимание на два решения тривиального квадратного уравнения и необходимость выбора второго (положительного) корня.

Сравните полученный результат с эталонным:

$$\gamma(\varepsilon) = \frac{8}{3}\pi m V(2m)^{1/2}\varepsilon^{3/2}.$$

В чём, по-вашему, преимущество последней записи с физической точки зрения?

Для Г-пространства одноатомного газа с N частицами объём V возведётся в степень N, а сфера импульсов будет многомерной. Используем формулу для объёма многомерной сферы, делаем необходимые подстановки и дифференцируем.

```
(%i9) Vn:(2*%pi*%e*r^2/n)^(n/2)$
(%i10) subst([n=3*N,r=sqrt(2*m*E)],%)*V^N;
```

(%o10) $$\frac{m^{\frac{3N}{2}} E^{\frac{3N}{2}} \%e^{\frac{3N}{2}} \pi^{\frac{3N}{2}} 2^{3N} V^N}{N^{\frac{3N}{2}} 3^{\frac{3N}{2}}}$$

```
(%i11) diff(%,E),radcan,factor;
```

(%o11) $$m^{\frac{3N}{2}} E^{\frac{3N}{2}-1} N^{1-\frac{3N}{2}} 3^{1-\frac{3N}{2}} \%e^{\frac{3N}{2}} \pi^{\frac{3N}{2}} 2^{3N-1} V^N$$

Сравните с эталонным ответом:

$$\Gamma(E) = \left(\frac{4\pi m e}{3N}\right)^{3N/2} V^N E^{3N/2},$$

$$g(E) = \frac{3N}{2}\left(\frac{4\pi m e}{3N}\right)^{3N/2} V^N E^{\frac{3N}{2}-1}.$$

Попробуйте заставить MAXIMA собрать в скобки множители с одинаковыми степенями.

Рассмотрим теперь фазовую траекторию маятника. Получаем уравнения Гамильтона.

```
(%i1) depends([phi,p],t)$ assume(g>0,l>0)$
(%i3) Ek:m/2*l^2*(diff(phi,t))^2$
(%i4) U:-m*g*l*cos(phi)$
(%i5) p=diff(Ek-U,diff(phi,t));
```

(%o5) $$p = l^2 m \left(\frac{d}{d\,t}\phi\right)$$

```
(%i6) solve(%,diff(phi,t));
```

(%o6) $$\left[\frac{d}{d\,t}\phi = \frac{p}{l^2 m}\right]$$

```
(%i7) H:subst(%,Ek+U);
```

(%o7) $$\frac{p^2}{2\,l^2 m} - g\,l\,m\cos(\phi)$$

```
(%i8) diff(phi,t)=diff(H,p); diff(p,t)=-diff(H,phi);
```

(%o8) $$\frac{d}{d\,t}\phi = \frac{p}{l^2 m}$$

(%o9) $$\frac{d}{d\,t}p = -g\,l\,m\sin(\phi)$$

Сводим их в одно и применяем приближение малых качаний $\sin\varphi \approx \varphi$.

```
(%i10) diff(%th(2),t);
```

(%o10) $$\frac{d^2}{d t^2}\phi=\frac{\frac{d}{d t}p}{l^2 m}$$

```
(%i11) subst([%th(2)],%);
```

(%o11) $$\frac{d^2}{d t^2}\phi=-\frac{g \sin(\phi)}{l}$$

```
(%i12) subst([sin(phi)=phi],%);
```

(%o12) $$\frac{d^2}{d t^2}\phi=-\frac{g\,\phi}{l}$$

```
(%i13) ode2(%,phi,t);
```

(%o13) $$\phi=\%k1 \sin\left(\frac{\sqrt{g}\,t}{\sqrt{l}}\right)+\%k2 \cos\left(\frac{\sqrt{g}\,t}{\sqrt{l}}\right)$$

```
(%i14) ic2(%,t=0,phi=phi0,diff(phi,t)=0);
```

(%o14) $$\phi=phi0 \cos\left(\frac{\sqrt{g}\,t}{\sqrt{l}}\right)$$

Исключаем время.

```
(%i15) solve(%th(7),p);
```

(%o15) $$[p=l^2 m\left(\frac{d}{d t}\phi\right)]$$

```
(%i16) subst([%th(2)],rhs(first(%)));
```

(%o16) $$l^2 m\left(\frac{d}{d t}\left(phi0 \cos\left(\frac{\sqrt{g}\,t}{\sqrt{l}}\right)\right)\right)$$

```
(%i17) p=l^2*m*(diff((phi0*cos((sqrt(g)*t)/sqrt(l))),t,1));
```

(%o17) $$p=-\sqrt{g}\,l^{3/2}\,m\,phi0 \sin\left(\frac{\sqrt{g}\,t}{\sqrt{l}}\right)$$

```
(%i18) solve(%th(4),t);
solve: using arc-trig functions to get a solution.
 Some solutions will be lost.
```

(%o18) $$[t=\frac{\sqrt{l}\,\operatorname{acos}\left(\frac{\phi}{phi0}\right)}{\sqrt{g}}]$$

```
(%i19) subst(%,%th(2)),factor;
```

(%o19) $$p=-\frac{\sqrt{g}\,l^{3/2}\,m\,phi0\sqrt{phi0^2-\phi^2}}{|phi0|}$$

```
(%i20) lhs(%)^2=rhs(%)^2;
```

(%o20) $$p^2=g\,l^3\,m^2\,(phi0^2-\phi^2)$$

Разберём пример нелинейной пружины [11]. Уравнения движения имеют вид

$$\dot{p} = -(\alpha q + \beta q^3);\ \dot{q} = p/m.$$

После интегрирования получаем не зависящий от времени гамильтониан

$$\frac{p^2}{2m} + \left[\frac{\alpha q^2}{2} + \frac{\beta q^4}{4}\right] = H(p_0, q_0).$$

Коэффициент α выбран положительным с тем, чтобы при небольших q сила была возвращающей, а коэффициент $\beta < 0$. Изобразим графически зависимость потенциальной энергии от положения частицы и соответствующие точки на фазовой плоскости для нескольких значений гамильтониана (рис.). Начальные значения гамильтониана соответствуют полной энергии системы (кинетическая + потенциальная). Если значение гамильтониана H больше максимального значения потенциальной энергии U, то импульс P всегда отличен от нуля; это ведёт к тому, что для гамильтониана H_U движение слева направо будет неограниченным. Для начальных значений амплитуд, расположенных внутри потенциальной ямы, значения начальной полной энергии H_S и H_L отвечают соответственно устойчивому движению и границе устойчивого движения.



Рис. Соответствие между диаграммой энергии (*а*) и диаграммой в фазовом пространстве (*б*) для нелинейного осциллятора.

(%i1) `pdot:-(a*q+b*q^3)$`

(%i2) `qdot:p/m$`

(%i3) `-integrate(%th(2),q)+integrate(%th(1),p);`

(%o3) $\frac{b\,q^4}{4}+\frac{a\,q^2}{2}+\frac{p^2}{2\,m}$

(%i4) `pickapart(%,1);`

(%t4) $\frac{b\,q^4}{4}$

(%t5) $\frac{a\,q^2}{2}$

(%t6) $\frac{p^2}{2\,m}$

(%o6) $\%t6+\%t5+\%t4$

(%i7) `%t4+%t5;`

(%o7) $\frac{b\,q^4}{4}+\frac{a\,q^2}{2}$

(%i8) `subst([a=1],%);`

(%o8) $\frac{b\,q^4}{4}+\frac{q^2}{2}$

(%i9) `solve(subst([q=1],%),b);`

(%o9) $[b=-2]$

(%i10) `subst(%,%th(2));`

(%o10) $\frac{q^2}{2}-\frac{q^4}{2}$

(%i11) `wxplot2d([%],[q,-1.1,1.1],[nticks,20])$`

(%t11)



5. Фазовое пространство (продолжение)

Задача 1 (3.2–3.7 из [7]). Определить фазовую траекторию тела массы m, которое движется в постоянном гравитационном поле из точки z_0 с начальной скоростью v_0, направленной вертикально вверх. Начертить эту траекторию в пространстве (z, p). Вычислить фазовый объём, ограниченный этой фазовой траекторией и осью p, если $z_0 = 0$.

Традиционно: из закона сохранения энергии следует уравнение параболы:

$$\Gamma = \int_{-p_0}^{p_0} z\,dp = \frac{2p_0^3}{3m^2 g}.$$

$$z = z_0 + \frac{p_0^2}{2m^2 g} - \frac{p^2}{2m^2 g}.$$

Компьютерное решение.

```
(%i1) m*g*(z-z0)=(p0^2-p^2)/(2*m)$
(%i2) solve(%,z);
```

(%o2) $\left[z=\frac{2\,g\,m^2\,z0+p0^2-p^2}{2\,g\,m^2}\right]$

```
(%i3) rhs(first(%)),expand;
```

(%o3) $z0+\frac{p0^2}{2\,g\,m^2}-\frac{p^2}{2\,g\,m^2}$

```
(%i4) integrate(%,p,-p0,p0);
```

(%o4) $\frac{2\,(3\,g\,m^2\,p0\,z0+p0^3)}{3\,g\,m^2}$

```
(%i5) subst([z0=0],%);
```

(%o5) $\frac{2\,p0^3}{3\,g\,m^2}$

Задача 2 (3.8 из [7]). Вывести уравнение фазовой траектории для частицы массы m, заряда e, которая движется под влиянием кулоновской силы притяжения к заряду e_1. Начальное расстояние между частицами r_0, а скорость $v_0 = 0$.

Традиционно: Из закона сохранения энергии

$$\int_{r_0}^{r} \frac{ee_1}{r^2}\,dr = \frac{p_0^2 - p^2}{2m} \qquad \text{и} \qquad p_0 = mv_0 = 0$$

следует $p = \sqrt{2mee_1\left[\frac{1}{r} - \frac{1}{r_0}\right]}$.

Компьютерное решение.

```
(%i1) assume(r>0,r0>0,r<r0)$
(%i2) U(r):=e*e1/r^2$
(%i3) integrate(U(r),r,r0,r)=integrate(p/m,p,p,p0);
```

$$(\%o3)\quad -e\,e1\left(\frac{1}{r}-\frac{1}{r0}\right)=\frac{\frac{p0^2}{2}-\frac{p^2}{2}}{m}$$

```
(%i4) subst([p0=0],%);
```

$$(\%o4)\quad -e\,e1\left(\frac{1}{r}-\frac{1}{r0}\right)=-\frac{p^2}{2\,m}$$

```
(%i5) solve(%,p);
```

$$(\%o5)\quad [p=-\frac{\sqrt{2}\sqrt{e\,e1\,m}\sqrt{r0-r}}{\sqrt{r}\sqrt{r0}},\ p=\frac{\sqrt{2}\sqrt{e\,e1\,m}\sqrt{r0-r}}{\sqrt{r}\sqrt{r0}}]$$

Задача 3 (3.9 из [7]). Вывести уравнение фазовой траектории для частицы массы m и заряда e, которая движется в постоянном электрическом поле E, направленном вдоль оси OX. Начальная координата частицы x_0, начальная скорость v_0, направленная параллельно оси OX.

Ответ: $x = x_0 + \frac{p^2-p_0^2}{2eEm}$.

Компьютерное решение.

```
(%i1) integrate(e*E,x,x0,x)=integrate(p/m,p,p0,p);
```

$$(\%o1)\quad e\,(x-x0)\,E=\frac{\frac{p^2}{2}-\frac{p0^2}{2}}{m}$$

```
(%i2) solve(%,x);
```

$$(\%o2)\quad [x=\frac{2\,e\,m\,x0\,E-p0^2+p^2}{2\,e\,m\,E}]$$

```
(%i3) rhs(first(%)),expand;
```

$$(\%o3)\quad -\frac{p0^2}{2\,e\,m\,E}+\frac{p^2}{2\,e\,m\,E}+x0$$

Задача 4 (3.10а из [7]). Найти на фазовой плоскости (q, v), где $V = \dot{q}$ – скорость, траектории свободной частицы, сила трения которой о среду пропорциональна скорости $f = -\gamma\dot{q}$. Вычислить также изменение фазового объёма dv dq.

Традиционно: решая уравнение движения $m\ddot{q} = -\gamma\dot{q}$ с начальными условиями $[\dot{q}(t=0) = v_0,\ q(t = 0) = q_0]$, найдём $q = q_0 + mv_0/\gamma - mv_0e^{-\gamma t/m}/\gamma$, $v = v_0e^{-\gamma t/m}$ и фазовую траекторию $q = q_0 + m(v_0 - v)/\gamma$. При движении системы по фазовой траектории из точки (q_0, v_0) в точку (q, v) её элементарный фазовый объём становится равным dv dq=$|D(v,q)/D(v_0,q_0)|$dv_0dq_0 = $e^{-\gamma t/m}$ dv_0 dq_0:то есть фазовый объём с течением времени уменьшается по обратной экспоненте. Теорема Лиувилля выполняется в отсутствие трения ($\gamma = 0$).

Компьютерное решение.

```
(%i1) depends(q,t)$ assume(gamma>0)$
(%i3) m*diff(q,t,2)=-gamma*diff(q,t)$
(%i4) ode2(%,q,t);
```

$$(\%o4)\quad q=\%k2\,\%e^{-\frac{t\Gamma}{m}}+\%k1$$

```
(%i5) ic2(%,t=0,q=q0,diff(q,t)=v0),expand;
```

$$(\%o5)\quad q=-\frac{m\,v0\,\%e^{-\frac{t\Gamma}{m}}}{\Gamma}+\frac{m\,v0}{\Gamma}+q0$$

```
(%i6) v=diff(rhs(%),t);
```

$$(\%o6)\quad v=v0\,\%e^{-\frac{t\Gamma}{m}}$$

```
(%i7) solve(%,t);
```

$$(\%o7)\quad [t=\frac{m\log\left(\frac{v0}{v}\right)}{\Gamma}]$$

```
(%i8) subst(%,%th(3));
```

$$(\%o8)\quad q=\frac{m\,v0}{\Gamma}-\frac{m\,v}{\Gamma}+q0$$

Задача 5 (3.10б из [7]). Найти на фазовой плоскости (q, v), где $v=\dot{q}$ – скорость, траектории и для линейного гармонического осциллятора с малым трением. Вычислить также изменение фазового объёма dvdq.

Традиционно. Уравнение движения $m\ddot{q}+m\omega_0^2m^2=-\gamma\dot{q}$ (см. задачу 4). Решение его с учётом малости γ ($4\omega_0^2m^2 >> \gamma^2$) начальных условий [$v(t=0)$ $= -q_0\omega_0$] : $q = q_0\, e^{-\gamma t/\omega} \cos \omega t$, $q = v = q_0 e^{-\dot{\gamma} t/\omega}\omega_0 \sin\omega_0 t$ описывает фазовую траекторию в виде эллипса с убывающими со временем по закону $e^{-\gamma t/\omega}$ полуосями:

$$\frac{q^2}{q_0^2e^{-2\gamma t/\omega}}+\frac{v^2}{q_0^2\omega_0^2e^{-2\gamma t/\omega}}=1.$$

Изменение фазового объёма равно $e^{-\gamma t/\omega}$.

Компьютерное решение.

```
(%i1) depends(q,t)$
(%i2) m*(diff(q,t,2)+omega0^2*q)=-gamma*diff(q,t)$
(%i3) ode2(%,q,t);
Is (Γ-2 m omega0)(Γ+2 m omega0) positive, negative or zero?n;
```

$$(\%o3)\quad q=\%e^{-\frac{t\Gamma}{2m}}\left(\%k1\sin\left(\frac{t\sqrt{4\,omega0^2-\frac{\Gamma^2}{m^2}}}{2}\right)+\%k2\cos\left(\frac{t\sqrt{4\,omega0^2-\frac{\Gamma^2}{m^2}}}{2}\right)\right)$$

```
(%i4) ic2(%,t=0,q=q0,diff(q,t)=-q0*omega0);
```

$$(\%o4)\quad q=\%e^{-\frac{t\Gamma}{2m}}\left(\frac{q0\,(\Gamma-2\,m\,omega0)\sin\left(\frac{t\sqrt{4\,omega0^2-\frac{\Gamma^2}{m^2}}}{2}\right)}{m\sqrt{4\,omega0^2-\frac{\Gamma^2}{m^2}}}+q0\cos\left(\frac{t\sqrt{4\,omega0^2-\frac{\Gamma^2}{m^2}}}{2}\right)\right)$$

Замечаем, что Γ стоит здесь, во-первых, под экспонентой, во-вторых, вычитается с величинами порядка $m\omega_0$, где, собственно, и проявляются его малые значения. Здесь целесообразно применить следующий подход–«трюк». Выделяем всё выражение, копируем в буфер, вставляем в следующую строку. Вручную «защищаем» Γ под экспонентой, переименовав, например, в “gama” (подобно тому, как химики-органики защищают реакционноспособные группы от нежелательных реакций). И делаем подстановку Γ = 0.

```
(%i5) %e^(-(t*gama)/(2*m))*((q0*(gamma-2*m*omega0)*
      sin((t*sqrt(4*omega0^2-gamma^2/m^2))/2))/
      (m*sqrt(4*omega0^2-gamma^2/m^2))+
      q0*cos((t*sqrt(4*omega0^2-gamma^2/m^2))/2))$
(%i6) subst([gamma=0],%),radcan;
```

$$(\%o6)\quad -\frac{\%e^{-\frac{gama\,t}{2m}}\left(omega0\;q0\sin(|omega0|\;t)-|omega0|\;q0\cos(|omega0|\;t)\right)}{|omega0|}$$

```
(%i7) v=diff(%,t);
```

$$(\%o7)\quad v=\frac{gama\;\%e^{-\frac{gama\,t}{2m}}\left(omega0\;q0\sin(|omega0|\;t)-|omega0|\;q0\cos(|omega0|\;t)\right)}{2\,m\,|omega0|}-\frac{\%e^{-\frac{gama\,t}{2m}}\left(omega0^2\;q0\sin(|omega0|\;t)+omega0\,|omega0|\;q0\cos(|omega0|\;t)\right)}{|omega0|}$$

```
(%i8) q=%th(2);
```

$$(\%o8)\quad q=-\frac{\%e^{-\frac{gama\,t}{2m}}\left(omega0\;q0\sin(|omega0|\;t)-|omega0|\;q0\cos(|omega0|\;t)\right)}{|omega0|}$$

Задача 6 (3.11 из [7]). Найти число состояний $\Omega(E)$ частиц газа, энергия которых связана с импульсом соотношением $\varepsilon = pc$, где c – константа.

Ответ: $\Omega(\varepsilon)d\varepsilon = \frac{1}{h^3}\frac{\partial\Gamma}{\partial\varepsilon}\,d\varepsilon$, $\Gamma = \frac{4\pi\varepsilon^3}{3c^3}$, $\Omega(\varepsilon)d\varepsilon = \frac{4\pi V\varepsilon^2}{c^3h^3}\,d\varepsilon$.

Компьютерное решение.

```
(%i1) E=p*c$
(%i2) gamma:V*(4/3)*%pi*p^3$
(%i3) solve(%th(2),p);
(%o3) [p=E/c]
(%i4) subst(%,%th(2));
(%o4) 4 π E^3 V / (3 c^3)
(%i5) diff(%,E)/h^3;
(%o5) 4 π E^2 V / (c^3 h^3)
```

Задача 7 (3.15 из [7]). Вычислить объём, который находится внутри гиперповерхности постоянной энергии, для пространственного ротатора. Обобщённые координаты ϕ и θ, обобщённые импульсы p_ϕ, p_θ. Указание: Энергия пространственного ротатора $H = \frac{p_\theta^2 + p_\varphi^2/sin^2\varphi}{2mr^2}$.

Традиционно: Задача сводится к вычислению интеграла $\Gamma = \int_{(\varphi)}\int_{(\theta)}\int_{(p_\varphi)}\int_{(p_\theta)} d\varphi\, d\theta\, dp_\varphi\, dp_\theta$ по области, ограниченной поверхностью $H =$ const. Пределы интегрирования $\phi \in [0, 2\pi]$, $\theta \in [0, \pi]$, $p_\varphi \in \left[-\sin\theta\sqrt{2mr^2H - p_0^2};\ \sin\theta\sqrt{2mr^2H - p_0^2}\right]$. После замены переменной $p_\theta = \sin u\sqrt{2mr^2H}$ имеем: $\Gamma = 4\pi mr^2H$.

Компьютерное решение.

```
(%i1) assume(m>0,H>0,r>0)$
(%i2) H=(ptheta^2+pphi^2/sin(phi)^2)/(2*m*r^2)$
(%i3) solve(%,pphi);
(%o3) [pphi=-sin(ϕ)√(2 m r^2 H-ptheta^2), pphi=sin(ϕ)√(2 m r^2 H-ptheta^2)]
(%i4) integrate(1,pphi,rhs(first(%)),rhs(second(%)));
(%o4) 2 sin(ϕ)√(2 m r^2 H-ptheta^2)
(%i5) solve(subst([pphi=0],%th(3)),ptheta);
(%o5) [ptheta=-√2 √m r √H, ptheta=√2 √m r √H]
(%i6) integrate(%th(2),ptheta,rhs(first(%)),rhs(second(%)));
(%o6) 2 π m sin(ϕ) r^2 H
(%i7) integrate(%,phi,0,%pi);
(%o7) 4 π m r^2 H
```

6. Теорема Лиувилля

Задача 1 (1.11 из [1]). Проверить справедливость теоремы Лиувилля для случая упругого соударения двух частиц, движущихся по одной прямой.

Традиционно. Обозначим координаты и импульсы частиц до и после соударения соответственно через q_i, p_i и q_i', p_i'. Из законов сохранения импульса $p_1 + p_2 = p_1' + p_2'$ и энергии

$$\frac{p_1^2}{2m_1} + \frac{p_2^2}{2m_2} = \frac{p_1'^2}{2m_1} + \frac{p_2'^2}{2m_2}$$

следует

$$p_1' = \frac{m_1 - m_2}{m_1 + m_2} p_1 + \frac{2m_1}{m_1 + m_2} p_2,$$

$$p_2' = -\frac{m_1 - m_2}{m_1 + m_2} p_2 + \frac{2m_2}{m_1 + m_2} p_1.$$

Теперь вычислим якобиан преобразования

$$D = \frac{\partial(q_1' q_2' p_1' p_2')}{\partial(q_1 q_2 p_1 p_2)}.$$

В силу соотношений $\partial p_i'/\partial q_j = 0$ и $\partial q_i'/\partial q_j = \delta_{ij}$:

$$D = \frac{\partial(p_1' p_2')}{\partial(p_1 p_2)} = \begin{vmatrix} \dfrac{m_1 - m_2}{m_1 + m_2} & \dfrac{2m_2}{m_1 + m_2} \\ \dfrac{2m_1}{m_1 + m_2} & -\dfrac{m_1 - m_2}{m_1 + m_2} \end{vmatrix} = \left(\frac{m_1 - m_2}{m_1 + m_2}\right)^2 - \frac{2m_1 m_2}{(m_1 + m_2)^2} = -1.$$

Очевидно, что $|D| = 1$, то есть фазовый объём сохраняется.

Компьютерное решение.

```
(%i1) p1a+p2a=p1b+p2b$
(%i2) p1a^2/m1+p2a^2/m2=p1b^2/m1+p2b^2/m2$
(%i3) solve([%,%th(2)],[p1b,p2b]);
(%o3) [[p1b=p1a,p2b=p2a],[p1b=(2 m1 p2a+(m1-m2) p1a)/(m2+m1),p2b=((m2-m1) p2a+2 m2 p1a)/(m2+m1)]]
(%i4) second(%);
(%o4) [p1b=(2 m1 p2a+(m1-m2) p1a)/(m2+m1),p2b=((m2-m1) p2a+2 m2 p1a)/(m2+m1)]
(%i5) jacobian([rhs(first(%)),rhs(second(%))],[p1a,p2a]);
(%o5) [ (m1-m2)/(m2+m1)   2 m1/(m2+m1) ]
      [ 2 m2/(m2+m1)      (m2-m1)/(m2+m1) ]
(%i6) determinant(%),ratsimp;
(%o6) -1
```

Задача 2 (1.12 из [1]). Проверить теорему Лиувилля для трёх гармонических осцилляторов:

$$x_1 = \sqrt{\frac{2\varepsilon}{m\omega^2}} \sin \omega t, \quad x_2 = \sqrt{\frac{2(\varepsilon + \Delta\varepsilon)}{m\omega^2}} \sin \omega t, \quad x_3 = \sqrt{\frac{2\varepsilon}{m\omega^2}} \sin(\omega t + \delta).$$

Традиционно. Координаты x_1, x_2, x_3 принимаем за обобщённые координаты. Тогда им соответствуют обобщённые импульсы

$$p_1 = \sqrt{2m\varepsilon}\cos\omega t,\; p_2 = \sqrt{2m(\varepsilon + \Delta\varepsilon)}\cos\omega t,\; p_3 = \sqrt{2m\varepsilon}\cos(\omega t + \delta).$$

Площадь треугольника выразим через координаты его вершин

$$s = \frac{1}{2}\begin{vmatrix} x_3 & p_3 & 1 \\ x_2 & p_2 & 1 \\ x_1 & p_1 & 1 \end{vmatrix} = \frac{1}{2}[(-p_3x_2 + p_2x_3) + (-p_1x_3 + p_3x_1) + (-p_2x_1 + p_1x_2)].$$

Подставляя значения обобщённых координат и импульсов:

$$s = m\sqrt{\varepsilon\Delta\varepsilon}\sin\delta,$$

что доказывает справедливость теоремы Лиувилля для трёх гармонических осцилляторов.

Компьютерное решение.

```
(%i1) assume(omega>0)$
(%i2) x1:sqrt(2*E/(m*omega^2))*sin(omega*t)$
(%i3) x2:sqrt(2*(E+dE)/(m*omega^2))*sin(omega*t)$
(%i4) x3:sqrt(2*E/(m*omega^2))*sin(omega*t+delta)$
(%i5) p1:m*diff(x1,t)$
(%i6) p2:m*diff(x2,t)$
(%i7) p3:m*diff(x3,t)$
(%i8) matrix([x1,p1,1],[x2,p2,1],[x3,p3,1]);
```

(%o8)
$$\begin{bmatrix} \dfrac{\sqrt{2}\sin(\omega t)\sqrt{\frac{E}{m}}}{\omega} & \sqrt{2}\,m\cos(\omega t)\sqrt{\frac{E}{m}} & 1 \\ \dfrac{\sqrt{2}\sin(\omega t)\sqrt{\frac{E+dE}{m}}}{\omega} & \sqrt{2}\,m\cos(\omega t)\sqrt{\frac{E+dE}{m}} & 1 \\ \dfrac{\sqrt{2}\sin(\omega t+\delta)\sqrt{\frac{E}{m}}}{\omega} & \sqrt{2}\,m\cos(\omega t+\delta)\sqrt{\frac{E}{m}} & 1 \end{bmatrix}$$

```
(%i9) determinant(%),radcan,trigreduce;
```

(%o9)
$$\frac{2\sin(\delta)E}{\omega} - \frac{2\sin(\delta)\sqrt{E}\sqrt{E+dE}}{\omega}$$

```
(%i10) factor(%);
```

(%o10)
$$-\frac{2\sin(\delta)(\sqrt{E}\sqrt{E+dE}-E)}{\omega}$$

Задача 3 (3.4 из [7]). Проверить теорему Лиувилля для системы из трёх материальных частиц массы m, движущихся в поле тяжести, если заданы их начальные координаты в фазовом пространстве: $A_1(p_0,z_0)$, $A_2(p_0+b,z_0)$, $A_3(p_0,z_0 + a)$.

Ответ. За время t вершины треугольника $A_1A_2A_3$ переместятся в точки фазового пространства $A_1'(p_1,z_1)$, $A_2'(p_2,z_2)$, $A_3'(p_3,z_3)$, где $p_1 = p_0 - mgt$, $z_1 = z_0 + p_0t/m - gt^2/2$, $p_2 = p_1$, $z_2 = z_1 + a$, $p_3 = p_1 + b$, $z_3 = z_1 + bt/m$. Из равенства площадей треугольников $A_1A_2A_3$ и $A_1'A_2'A_3'$ следует справедливость теоремы Лиувилля для данного случая.

Компьютерное решение.

```
(%i1) matrix([z0,p0,1],[z0,p0+b,1],[z0+a,p0,1]);
```

$$(\%o1)\quad \begin{bmatrix} z0 & p0 & 1 \\ z0 & p0+b & 1 \\ z0+a & p0 & 1 \end{bmatrix}$$

```
(%i2) determinant(%),ratsimp;
(%o2) -a b
(%i3) p1:p0-m*g*t$
(%i4) z1:z0+p0*t/m-g*t^2/2$
(%i5) p2:p1$
(%i6) z2:z1+a$
(%i7) p3:p1+b$
(%i8) z3:z1+b*t/m$
(%i9) matrix([z1,p1,1],[z2,p2,1],[z3,p3,1]);
```

$$(\%o9)\quad \begin{bmatrix} z0-\frac{g\,t^2}{2}+\frac{p0\,t}{m} & p0-g\,m\,t & 1 \\ z0-\frac{g\,t^2}{2}+\frac{p0\,t}{m}+a & p0-g\,m\,t & 1 \\ z0-\frac{g\,t^2}{2}+\frac{p0\,t}{m}+\frac{b\,t}{m} & -g\,m\,t+p0+b & 1 \end{bmatrix}$$

```
(%i10) determinant(%),ratsimp;
(%o10) a b
```

Задача 4 (Пример 9 из [8]). Проверить теорему Лиувилля для линейного гармонического осциллятора, движущегося с малым трением ($F = -\kappa\dot{q}$, $\omega \gg \kappa/m$).

Традиционно. Запишем уравнение движения

$$m\ddot{q} = -kq - \kappa\dot{q} \text{ или } \ddot{q} + \frac{\kappa}{m}\dot{q} + \omega^2 q = 0,$$

где $\omega = \sqrt{k/m}$. Решение этого дифференциального уравнения ищем в виде $q(t) = C\exp(\alpha t)$.

Получаем характеристическое уравнение $\alpha^2+(\kappa/m)\alpha+\omega^2=0$, для которого находим корни:

$$\alpha = -\frac{\kappa}{2m} \pm \sqrt{\frac{\kappa^2}{4m^2} - \omega^2} \approx -\frac{\kappa}{2m} \pm i\omega, \text{ так как } \omega \gg \kappa/m.$$

Следовательно, решение уравнения движения можно записать в виде:

$$q(t) = e^{-\kappa t/2m}(A \sin \omega t + B \cos \omega t),$$

$$p(t) = m e^{-\kappa t/2m}\left[(A\omega \cos \omega t - B\,\omega \sin \omega t) - \frac{\kappa}{2m}(A \sin \omega t + B \cos \omega t)\right].$$

Учитывая, что трение – мало, вторым слагаемым в $p(t)$ можно пренебречь. По начальным условиям $q(t) = q_0$ и $p(t) = p_0$ определяем константы: $A = p_0/(m\omega)$, $B = q_0$. И далее записываем и вычисляем якобиан.

Компьютерное решение.

```
(%i1) depends(q,t)$ assume(m>0)$
(%i3) diff(q,t,2)+kappa*diff(q,t)/m+omega^2*q=0$
(%i4) ode2(%,q,t)$
Is (2 m ω-κ)(2 m ω+κ) positive, negative or zero?p;
(%i5) ic2(%,t=0,q=q0,diff(q,t)=p0/m);
```

$$\text{(\%o5)}\quad q = \%e^{-\frac{\kappa t}{2m}}\left(\frac{(\kappa\, q0 + 2\, p0)\sin\left(\frac{\sqrt{4\omega^2 - \frac{\kappa^2}{m^2}}\, t}{2}\right)}{m\sqrt{4\omega^2 - \frac{\kappa^2}{m^2}}} + q0 \cos\left(\frac{\sqrt{4\omega^2 - \frac{\kappa^2}{m^2}}\, t}{2}\right)\right)$$

```
(%i6) p:m*diff(rhs(%),t),radcan$
(%i7) jacobian([rhs(%o5),%],[q0,p0])$
(%i8) determinant(%),radcan,trigreduce;
```

$$\text{(\%o8)}\quad \%e^{-\frac{\kappa t}{m}}$$

Задача 5 ([9]). Проверить теорему Лиувилля для осциллятора с координатой $x(t) = ae^{-\gamma t}\cos \omega t + be^{-\gamma t}\sin \omega t$, причем $\gamma << \omega$.

Компьютерное решение.

```
(%i1) x:a*exp(-gamma*t)*cos(omega*t)+b*exp(-gamma*t)*sin(omega*t)$
(%i2) p:m*diff(x,t)$
(%i3) subst([t=0],%th(2));
(%o3) a
(%i4) subst([t=0],p);
(%o4) m(b ω - a Γ)
(%i5) solve([%th(2)=x0,%=p0],[a,b]);
                 m x0 Γ + p0
(%o5) [[a=x0, b= ----------- ]]
                    m ω
(%i6) subst(first(%),x);
      sin(ω t)(m x0 Γ + p0) %e^(-t Γ)
(%o6) ------------------------------- + cos(ω t) x0 %e^(-t Γ)
                  m ω
(%i7) subst(first(%th(2)),p)$
(%i8) jacobian([%th(2),%],[x0,p0])$
(%i9) determinant(%),trigreduce;
(%o9) %e^(-2 t Γ)
```

Задача 6 (10 из [10]). Определить фазовые траектории четырёх произвольных точек статистической системы, совершающих одномерное движение в постоянном поле тяжести. Проверить выполнимость теоремы Лиувилля.

Решение из оригинала. Для каждой материальной точки, совершающей одномерное движение, фазовое пространство – двумерное. Составим уравнения движения выбранных 4-х материальных точек. Для упрощения решения расположим их на двухмерной фазовой плоскости по вершинам произвольного прямоугольника.

В обычном геометрическом пространстве движение частиц в поле тяжести оказывается ускоренным согласно уравнениям

$$q = q_0 + \frac{p_0}{m}t + \frac{gt^2}{2}, \qquad p = p_0 + mgt.$$

Решим совместно эту систему уравнений: исключим t, согласно второму уравнению $t = (p - p_0)/mg$. После элементарных преобразований уравнение фазовой траектории:

$$(q - q_0)mg = \frac{p^2 - p_0^2}{2m}.$$

Вид этого уравнения позволяет утверждать, что фазовая траектория каждой материальной точки системы — это парабола с вершиной на оси Oq.

Для проверки выполнимости теоремы Лиувилля определим начальный и промежуточный фазовые объёмы, занимаемые избранными 4-мя материальными точками. Как видно из рисунка, первоначальный фазовый объём равен $\Delta\Gamma = \Delta q_0 \cdot \Delta p_0$. Определим начальные фазовые координаты всех 4-х материальных точек:

1	q_0	p_0
2	$q_0 + \Delta q_0$	p_0
3	q_0	$p_0 + \Delta p_0$
4	$q_0 + \Delta q_0$	$p_0 + \Delta p_0$

Через промежуток времени t обобщённые координаты будут:

1	$q_1 = q_0 + \frac{p_0 t}{m} + \frac{g t^2}{2}$	$p_1 = p_0 + mgt$
2	$q_2 = q_0 + \Delta q_0 + \frac{p_0 t}{m} + \frac{g t^2}{2}$	$p_2 = p_0 + mgt$
3	$q_3 = q_0 + \frac{p_0 t}{m} + \frac{\Delta p_0 t}{m} + \frac{g t^2}{2}$	$p_3 = p_0 + \Delta p_0 + mgt$
4	$q_4 = q_0 + \Delta q_0 + \frac{p_0 t}{m} + \frac{\Delta p_0 t}{m} + \frac{g t^2}{2}$	$P_4 = p_0 + \Delta p_0 + mgt$

Убедимся, что в новом положении четыре точки располагаются по вершинам параллелограмма. Действительно, разности $q_4 – q_3 = q_2 – q_1 = \Delta q_0$.

Соответственно, $p_4 – p_2 = p_2 – p_1 = 0$.



По одному из признаков параллелограмма (противоположные стороны равны по длине и параллельны) утверждаем, что точки располагаются по вершинам параллелограмма с площадью $S = \Delta q_0 \cdot \Delta p_0 = \Delta\Gamma$.

Тем самым доказали выполнение теоремы Лиувилля.

<u>Компьютерное решение</u>.

```
(%i1) assume(dq0>0,dp0>0)$
(%i2) q=q0+p0*t/m+g*t^2/2$
(%i3) p=p0+m*g*t$
(%i4) solve(%,t);
```

(%o4) $[t=-\frac{p0-p}{g\,m}]$

```
(%i5) subst(%,%th(3)),expand;
```

(%o5) $q=q0-\frac{p0^2}{2\,g\,m^2}+\frac{p^2}{2\,g\,m^2}$

```
(%i6) q(q0,p0):=q0+p0*t/m+g*t^2/2$
(%i7) p(p0):=p0+m*g*t$
(%i8) q1:q(q0,p0)$ p1:p(p0)$
(%i10) q2:q(q0+dq0,p0)$ p2:p(p0)$
(%i12) q3:q(q0,p0+dp0)$ p3:p(p0+dp0)$
(%i14) q4:q(q0+dq0,p0+dp0)$ p4:p(p0+dp0)$
(%i16) [q4-q3,q2-q1];
```

(%o16) $[dq0,dq0]$

```
(%i17) [p4-p3,p2-p1,p3-p1,p4-p2];
```

(%o17) $[0,0,dp0,dp0]$

```
(%i18) (abs(determinant(matrix([q1,p1,1],[q3,p3,1],[q4,p4,1])))+
       abs(determinant(matrix([q4,p4,1],[q2,p2,1],[q1,p1,1]))))/2,ratsimp;
```

(%o18) $dp0\,dq0$

Теперь решим эту задачу «вслепую» (подавляя вывод громоздких промежуточных выражений и выводя на экран только конечное решение) для *любых* значений начальных координат и импульсов.

```
(%i19) q1:q(q01,p01)$ p1:p(p01)$
       q2:q(q02,p02)$ p2:p(p02)$
       q3:q(q03,p03)$ p3:p(p03)$
       q4:q(q04,p04)$ p4:p(p04)$
(%i27) S1:(abs(determinant(matrix([q01,p01,1],[q03,p03,1],[q04,p04,1])))+
       abs(determinant(matrix([q04,p04,1],[q02,p02,1],[q01,p01,1]))))/2$
(%i28) S2:(abs(determinant(matrix([q1,p1,1],[q3,p3,1],[q4,p4,1])))+
       abs(determinant(matrix([q4,p4,1],[q2,p2,1],[q1,p1,1]))))/2$
(%i29) S2-S1,ratsimp;
(%o29) 0
```

7. Статистический интеграл и каноническое распределение

Вывод термодинамических величин из статистического интеграла.

```
(%i1) depends([Z,lnZ,F],[T,V])$
(%i2) F:-k*T*log(Z)$
(%i3) S:-diff(F,T);
```

$$(\%o3)\quad \frac{k\,T\left(\frac{d}{d\,T}\,Z\right)}{Z}+k\log(Z)$$

```
(%i4) p:-diff(F,V);
```

$$(\%o4)\quad \frac{k\,T\left(\frac{d}{d\,V}\,Z\right)}{Z}$$

```
(%i5) U:F+T*S,expand;
```

$$(\%o5)\quad \frac{k\,T^2\left(\frac{d}{d\,T}\,Z\right)}{Z}$$

```
(%i6) H:U+p*V,expand;
```

$$(\%o6)\quad \frac{k\,T\,V\left(\frac{d}{d\,V}\,Z\right)}{Z}+\frac{k\,T^2\left(\frac{d}{d\,T}\,Z\right)}{Z}$$

```
(%i7) G:F+p*V;
```

$$(\%o7)\quad \frac{k\,T\,V\left(\frac{d}{d\,V}\,Z\right)}{Z}-k\,T\log(Z)$$

```
(%i8) Cv:diff(U,T);
```

$$(\%o8)\quad \frac{k\,T^2\left(\frac{d^2}{d\,T^2}\,Z\right)}{Z}-\frac{k\,T^2\left(\frac{d}{d\,T}\,Z\right)^2}{Z^2}+\frac{2\,k\,T\left(\frac{d}{d\,T}\,Z\right)}{Z}$$

Здесь логарифм был раскрыт. Но выражения выглядят более компактно с нераскрытым логарифмом. Поэтому сделаем так.

```
(%i9) F:-k*T*lnZ$
(%i10) S:-diff(F,T),factor;
```

$$(\%o10)\quad k\left(\left(\frac{d}{d\,T}\,lnZ\right)T+lnZ\right)$$

```
(%i11) p:-diff(F,V);
```

$$(\%o11)\quad k\left(\frac{d}{d\,V}\,lnZ\right)T$$

```
(%i12) U:F+T*S,expand;
```

$$(\%o12)\quad k\left(\frac{d}{d\,T}\,lnZ\right)T^2$$

```
(%i13) H:U+p*V,expand;
```

$$(\%o13)\quad k\left(\frac{d}{d\,V}\,lnZ\right)T\,V+k\left(\frac{d}{d\,T}\,lnZ\right)T^2$$

```
(%i14) G:F+p*V,factor;
```

$$(\%o14)\quad k\,T\left(\left(\frac{d}{d\,V}\,lnZ\right)V-lnZ\right)$$

```
(%i15) Cv:diff(U,T);
```

$$(\%o15)\quad k\left(\frac{d^2}{d\,T^2}\,lnZ\right)T^2+2\,k\left(\frac{d}{d\,T}\,lnZ\right)T$$

Выражения для канонического распределения.

```
(%i1) assume(N>0,theta>0)$ declare(N,integer)$ load(stirling)$
(%i4) Vn:(2*%pi*%e*r^2/n)^(n/2)$
(%i5) subst([n=3*N,r=sqrt(2*m*E)],%)*V^N;
```

$$(\%o5)\quad \frac{(m\,E)^{\frac{3N}{2}}\,\%e^{\frac{3N}{2}}\,\pi^{\frac{3N}{2}}\,2^{3N}\,V^N}{N^{\frac{3N}{2}}\,3^{\frac{3N}{2}}}$$

```
(%i6) g:diff(%,E)$
(%i7) c:g/(gamma(N+1)*h^(3*N));
```

$$(\%o7)\quad \frac{(m\,E)^{\frac{3N}{2}}\,N^{1-\frac{3N}{2}}\,3^{1-\frac{3N}{2}}\,\%e^{\frac{3N}{2}}\,\pi^{\frac{3N}{2}}\,2^{3N-1}\,V^N}{h^{3N}\,E\,\Gamma(N+1)}$$

```
(%i8) c*exp((F-E)/theta)$
(%i9) diff(%,E)$
(%i10) solve(%,E);
```

$$(\%o10)\quad [E=\frac{3\,\theta\,N}{2}-\theta,\ E=0]$$

```
(%i11) integrate(%th(3),E,0,inf);
```

Is $\frac{3N}{2}$ *an integer?* y;

$$(\%o11)\quad \frac{m^{\frac{3N}{2}}\,\theta^{\frac{3N}{2}}\,N^{1-\frac{3N}{2}}\,3^{1-\frac{3N}{2}}\,\pi^{\frac{3N}{2}}\,2^{3N-1}\,\%e^{\frac{3\theta N+2F}{2\theta}}\,\Gamma\left(\frac{3N}{2}\right)V^N}{h^{3N}\,\Gamma(N+1)}$$

```
(%i12) f:stirling(%th(4)/%,1),expand,factor;
```

$$(\%o12)\quad \frac{(m\,E)^{\frac{3N}{2}}\,N^{1/2-\frac{3N}{2}}\,3^{1/2-\frac{3N}{2}}\,2^{\frac{3N}{2}-1}\,\%e^{\frac{3N}{2}-\frac{E}{\theta}}}{\sqrt{\pi}\,m^{\frac{3N}{2}}\,\theta^{\frac{3N}{2}}\,E}$$

```
(%i13) subst(first(%th(3)),f),radcan,factor;
```

$$(\%o13)\quad \frac{\%e\,N^{1/2-\frac{3N}{2}}\,(3N-2)^{\frac{3N}{2}-1}\,3^{1/2-\frac{3N}{2}}}{\sqrt{\pi}\,\theta}$$

Здесь не проходит напрашивающийся предел $N\to\infty$, поэтому вручную копируем выражение и убираем конечные числа, складывающиеся/вычитающиеся с N.

```
(%i14) (%e*N^(-(3*N)/2)*(3*N)^((3*N)/2)*3^(-(3*N)/2))/(sqrt(%pi)*theta)$
(%i15) 1/%;
```

$$(\%o15)\quad \%e^{-1}\sqrt{\pi}\,\theta$$

8. Модель идеального газа (поступательное движение)

Распределение молекул по компонентам импульса.

```
(%i1) assume(k>0,m>0,T>0)$
(%i2) exp(-px^2/(2*m*k*T))$
(%i3) integrate(%,px,-inf,inf);
```

(%o3) $\sqrt{2}\,\sqrt{\pi}\,\sqrt{k}\,\sqrt{m}\,\sqrt{T}$

```
(%i4) rho_px:%th(2)/%;
```

(%o4) $\dfrac{\%e^{-\frac{px^2}{2\,k\,m\,T}}}{\sqrt{2}\,\sqrt{\pi}\,\sqrt{k}\,\sqrt{m}\,\sqrt{T}}$

Распределение молекул по компонентам скорости.

```
(%i5) f:subst([px=m*vx],rho_px)*diff(m*vx,vx);
```

(%o5) $\dfrac{\sqrt{m}\,\%e^{-\frac{m\,vx^2}{2\,k\,T}}}{\sqrt{2}\,\sqrt{\pi}\,\sqrt{k}\,\sqrt{T}}$

Распределение молекул по модулю скорости.

```
(%i6) Vx:v*sin(theta)*cos(phi)$
(%i7) Vy:v*sin(theta)*sin(phi)$
(%i8) Vz:v*cos(theta)$
(%i9) trigsimp(determinant(jacobian([Vx,Vy,Vz],[v,theta,phi])));
```

(%o9) $\sin(\theta)\,v^2$

```
(%i10) %*exp(-m*v^2/(2*k*T))$
(%i11) integrate(%,theta,0,%pi);
```

(%o11) $2\,v^2\,\%e^{-\frac{m\,v^2}{2\,k\,T}}$

```
(%i12) integrate(%,phi,0,2*%pi);
```

(%o12) $4\,\pi\,v^2\,\%e^{-\frac{m\,v^2}{2\,k\,T}}$

Нормируем:

```
(%i13) integrate(%,v,0,inf);
```

(%o13) $\dfrac{2^{3/2}\,\pi^{3/2}\,k^{3/2}\,T^{3/2}}{m^{3/2}}$

```
(%i14) F:%th(2)/%;
```

(%o14) $\dfrac{\sqrt{2}\,m^{3/2}\,v^2\,\%e^{-\frac{m\,v^2}{2\,k\,T}}}{\sqrt{\pi}\,k^{3/2}\,T^{3/2}}$

Находим максимум:

(%i15) `diff(F,v);`

(%o15) $$\frac{2^{3/2}\, m^{3/2}\, v\, \%e^{-\frac{m\, v^2}{2\, k\, T}}}{\sqrt{\pi}\, k^{3/2}\, T^{3/2}}-\frac{\sqrt{2}\, m^{5/2}\, v^3\, \%e^{-\frac{m\, v^2}{2\, k\, T}}}{\sqrt{\pi}\, k^{5/2}\, T^{5/2}}$$

(%i16) `solve(%,v);`

(%o16) $$[v=-\frac{\sqrt{2}\sqrt{k}\sqrt{T}}{\sqrt{m}},\ v=\frac{\sqrt{2}\sqrt{k}\sqrt{T}}{\sqrt{m}},\ v=0]$$

(%i17) `vm=rhs(second(%)); assume(vm>0)$`

(%o17) $$vm=\frac{\sqrt{2}\sqrt{k}\sqrt{T}}{\sqrt{m}}$$

(%i19) `solve(%th(2),T);`

(%o19) $$[T=\frac{m\, vm^2}{2\, k}]$$

(%i20) `subst(%,F);`

(%o20) $$\frac{4\, v^2\, \%e^{-\frac{v^2}{vm^2}}}{\sqrt{\pi}\, vm^3}$$

(%i21) `subst([vm=1,v=vtilde],%);`

(%o21) $$\frac{4\, vtilde^2\, \%e^{-vtilde^2}}{\sqrt{\pi}}$$

Распределение молекул по энергиям поступательного движения:

(%i22) `e=m*v^2/2$ solve(%,v);`

(%o23) $$[v=-\frac{\sqrt{2}\sqrt{e}}{\sqrt{m}},\ v=\frac{\sqrt{2}\sqrt{e}}{\sqrt{m}}]$$

(%i24) `subst(second(%),F)*diff(rhs(second(%)),e);`

(%o24) $$\frac{2\sqrt{e}\, \%e^{-\frac{e}{k\, T}}}{\sqrt{\pi}\, k^{3/2}\, T^{3/2}}$$

Среднее значение v_x:

(%i25) `integrate(vx*f,vx,0,inf);`

(%o25) $$\frac{\sqrt{k}\sqrt{T}}{\sqrt{2}\sqrt{\pi}\sqrt{m}}$$

(%i26) `%/rhs(%th(9));`

(%o26) $$\frac{1}{2\sqrt{\pi}}$$

Среднее значения модуля скорости молекул:

```
(%i27) integrate(v*F,v,0,inf);
```

$$(\%o27)\quad \frac{2^{3/2}\sqrt{k}\sqrt{T}}{\sqrt{\pi}\sqrt{m}}$$

```
(%i28) %/rhs(%th(11));
```

$$(\%o28)\quad \frac{2}{\sqrt{\pi}}$$

Среднее значение v_x^2:

```
(%i29) integrate(vx^2*f,vx,-inf,inf);
```

$$(\%o29)\quad \frac{k\,T}{m}$$

Среднеквадратичная скорость молекул:

```
(%i30) sqrt(3*%);
```

$$(\%o30)\quad \frac{\sqrt{3}\sqrt{k}\sqrt{T}}{\sqrt{m}}$$

Среднее значение кинетической энергии поступательного движения молекул:

```
(%i31) m/2*integrate(v^2*F,v,0,inf);
```

$$(\%o31)\quad \frac{3\,k\,T}{2}$$

Среднее число ударов молекул о стенки сосуда:

```
(%i32) N/V*f*vx;
```

$$(\%o32)\quad \frac{\sqrt{m}\,vx\,N\,\%e^{-\frac{m\,vx^2}{2\,k\,T}}}{\sqrt{2}\sqrt{\pi}\sqrt{k}\sqrt{T}\,V}$$

```
(%i33) integrate(%,vx,0,inf);
```

$$(\%o33)\quad \frac{\sqrt{k}\,N\sqrt{T}}{\sqrt{2}\sqrt{\pi}\sqrt{m}\,V}$$

Давление газа:

```
(%i34) 2*m*integrate(vx*%th(2),vx,0,inf);
```

$$(\%o34)\quad \frac{k\,N\,T}{V}$$

Альтернативный расчёт числа ударов молекул о стенки сосуда:

```
(%i35) N/V*integrate(f,vx,x/t,inf);
```

$$(\%o35)\quad \frac{\sqrt{m}\,N\left(\frac{\sqrt{\pi}\sqrt{k}\sqrt{T}}{\sqrt{2}\sqrt{m}}-\frac{\sqrt{\pi}\sqrt{k}\,\mathrm{erf}\left(\frac{\sqrt{m}\,x}{\sqrt{2}\sqrt{k}\,t\sqrt{T}}\right)\sqrt{T}}{\sqrt{2}\sqrt{m}}\right)}{\sqrt{2}\sqrt{\pi}\sqrt{k}\sqrt{T}\,V}$$

```
(%i36) integrate(%,x,0,inf),ratsimp;
```

$$(\%o36)\quad -\frac{N\int_0^{\infty}\sqrt{2}\,\mathrm{erf}\left(\frac{\sqrt{m}\,x}{\sqrt{2}\sqrt{k}\,t\sqrt{T}}\right)-\sqrt{2}\,dx}{2^{3/2}\,V}$$

```
(%i37) pickapart(%,7);
```

$$(\%t37)\quad \frac{\sqrt{m}\,x}{\sqrt{2}\sqrt{k}\,t\sqrt{T}}$$

$$(\%o37)\quad -\frac{\int_0^{\infty}\sqrt{2}\,\mathrm{erf}(\%t37)-\sqrt{2}\,dx\;N}{2^{3/2}\,V}$$

Вручную копируем и заменяем %t37 на Y:

```
(%i38) (integrate(sqrt(2)*erf(Y)-sqrt(2),Y,0,inf)*N)/(2^(3/2)*V)/diff(%t37,x);
```

$$(\%o38)\quad -\frac{\sqrt{k}\,t\,N\sqrt{T}}{\sqrt{2}\sqrt{\pi}\sqrt{m}\,V}$$

```
(%i39) subst([t=1],%);
```

$$(\%o39)\quad -\frac{\sqrt{k}\,N\sqrt{T}}{\sqrt{2}\sqrt{\pi}\sqrt{m}\,V}$$

Альтернативный расчёт давления газа:

```
(%i40) assume(t>0)$
(%i41) N/V*integrate(2*m*vx*f,vx,x/t,inf);
```

$$(\%o41)\quad \frac{\sqrt{2}\sqrt{k}\sqrt{m}\,N\sqrt{T}\,\%e^{-\frac{m\,x^2}{2\,k\,t^2\,T}}}{\sqrt{\pi}\,V}$$

```
(%i42) integrate(%,x,0,inf);
```

$$(\%o42)\quad \frac{k\,t\,N\,T}{V}$$

```
(%i43) subst([t=1],%);
```

$$(\%o43)\quad \frac{k\,N\,T}{V}$$

9. Модель идеального газа (вращательное движение)

В выражении для кинетической энергии молекул переходим от обобщённых координат к обобщённым импульсам молекул:

```
(%i1) depends([x,y,z],t)$
(%i2) Ek:m/2*(diff(x,t)^2+diff(y,t)^2+diff(z,t)^2)$
(%i3) px=diff(Ek,diff(x,t));
```

$$(\%o3)\quad px=m\left(\frac{d}{d\,t}\,x\right)$$

```
(%i4) py=diff(Ek,diff(y,t))$
(%i5) pz=diff(Ek,diff(z,t))$
(%i6) solve(%th(3),diff(x,t));
```

$$(\%o6)\quad [\frac{d}{d\,t}\,x=\frac{px}{m}]$$

```
(%i7) solve(%th(3),diff(y,t))$
(%i8) solve(%th(3),diff(z,t))$
(%i9) subst(append(%,%th(2),%th(3)),Ek),ratsimp;
```

$$(\%o9)\quad \frac{pz^2+py^2+px^2}{2\,m}$$

Теперь вводим вращательные координаты и кинетическую энергию вращательного движения молекул:

```
(%i10) depends([theta,phi,r],t)$
(%i11) Erot:mu*r^2/2*(diff(theta,t)^2+sin(theta)^2*diff(phi,t)^2)$
(%i12) ptheta=diff(Erot,diff(theta,t));
```

$$(\%o12)\quad ptheta=\mu\,r^2\left(\frac{d}{d\,t}\,\theta\right)$$

```
(%i13) pphi=diff(Erot,diff(phi,t));
```

$$(\%o13)\quad pphi=\mu\left(\frac{d}{d\,t}\,\phi\right)r^2\,\sin(\theta)^2$$

```
(%i14) solve(%th(2),diff(theta,t));
```

$$(\%o14)\quad [\frac{d}{d\,t}\,\theta=\frac{ptheta}{\mu\,r^2}]$$

```
(%i15) solve(%th(2),diff(phi,t));
```

$$(\%o15)\quad [\frac{d}{d\,t}\,\phi=\frac{pphi}{\mu\,r^2\,\sin(\theta)^2}]$$

```
(%i16) subst(append(%,%th(2)),Erot),expand;
```

$$(\%o16)\quad \frac{pphi^2}{2\,\mu\,r^2\,\sin(\theta)^2}+\frac{ptheta^2}{2\,\mu\,r^2}$$

Выводим выражения для термодинамических функций модели одноатомного идеального газа: кинетической энергии поступательного движения и внутренних степеней свободы атомов, отражаемых величиной Q_i).

Статистический интеграл:

```
(%i1) assume(k>0,m>0,T>0)$ depends(Qi,T)$
(%i3) exp(-px^2/(2*m*k*T))$
(%i4) (integrate(%,px,-inf,inf)/h)^3*V$
(%i5) Z:(%*%e/N)^N*Qi^N;
```

$$(\%o5)\quad k^{\frac{3N}{2}}\, m^{\frac{3N}{2}}\, Qi^N\, \%e^N\, 2^{\frac{3N}{2}}\, \pi^{\frac{3N}{2}}\, T^{\frac{3N}{2}} \left(\frac{V}{h^3 N}\right)^N$$

Свободная энергия Гельмгольца:

```
(%i6) F:-k*T*log(Z), logexpand=super$
(%i7) %,logcontract$ expand(%);
```

$$(\%o8)\quad -\frac{k\,N\,T\log\left(\frac{8\,\pi^3\,k^3\,m^3\,Qi^2\,T^3\,V^2}{h^6\,N^2}\right)}{2}-k\,N\,T$$

Энтропия:

```
(%i9) S:-diff(F,T), logexpand=super$
(%i10) %,logcontract$ expand(%);
```

$$(\%o11)\quad \frac{k\,N\log\left(\frac{8\,\pi^3\,k^3\,m^3\,Qi^2\,T^3\,V^2}{h^6\,N^2}\right)}{2}+\frac{k\left(\frac{d}{d\,T}\,Qi\right)N\,T}{Qi}+\frac{5\,k\,N}{2}$$

Давление (обратите внимание — внутренние степени свободы не играют роли):

```
(%i12) p:-diff(F,V), logexpand=super;
```

$$(\%o12)\quad \frac{k\,N\,T}{V}$$

Свободная энергия Гиббса:

```
(%i13) G:F+p*V,logcontract;
```

$$(\%o13)\quad -\frac{k\,N\,T\log\left(\frac{8\,\pi^3\,k^3\,m^3\,Qi^2\,T^3\,V^2}{h^6\,N^2}\right)}{2}$$

Внутренняя энергия и энтальпия:

```
(%i14) U:F+T*S,logcontract$ expand(%);
```

$$(\%o15)\quad \frac{k\left(\frac{d}{d\,T}\,Qi\right)N\,T^2}{Qi}+\frac{3\,k\,N\,T}{2}$$

```
(%i16) H:U+p*V,logcontract$ expand(%);
```

$$(\%o17)\quad \frac{k\left(\frac{d}{d\,T}\,Qi\right)N\,T^2}{Qi}+\frac{5\,k\,N\,T}{2}$$

Теплоёмкости (при постоянном объёме и давлении):

```
(%i18) Cv:diff(U,T),expand;
```

$$(\%o18)\quad \frac{k\left(\frac{d^2}{d\,T^2}\,Qi\right)N\,T^2}{Qi}-\frac{k\left(\frac{d}{d\,T}\,Qi\right)^2 N\,T^2}{Qi^2}+\frac{2\,k\left(\frac{d}{d\,T}\,Qi\right)N\,T}{Qi}+\frac{3\,k\,N}{2}$$

```
(%i19) Cp:Cv+N*k,expand;
```

$$(\%o19)\quad \frac{k\left(\frac{d^2}{d\,T^2}\,Qi\right)N\,T^2}{Qi}-\frac{k\left(\frac{d}{d\,T}\,Qi\right)^2 N\,T^2}{Qi^2}+\frac{2\,k\left(\frac{d}{d\,T}\,Qi\right)N\,T}{Qi}+\frac{5\,k\,N}{2}$$

Сравните с выражениями из (любого) учебника физической химии.

10. Элементы квантовой статистики

Пробуем решить уравнение Шредингера для одной координаты частицы в прямоугольном потенциальном ящике

$$\frac{\partial^2 \psi_x(x)}{\partial x^2} + k_x^2\, \psi_x(x) = 0.$$

Вместо нуля задаём некую переменную *a* с намерением устремить её к нулю.

```
(%i1) depends(psi,x)$ assume(k>0,L>0)$
(%i3) declare([n,nx,ny,nz],integer)$
(%i4) diff(psi,x,2)+k^2*psi=0$
(%i5) ode2(%,psi,x);
```

(%o5) $\Psi = \%k1\,\sin(k\,x) + \%k2\,\cos(k\,x)$

```
(%i6) bc2(%,x=0,psi=0,x=L,psi=a);
```

(%o6) $\Psi = \dfrac{a\,\sin(k\,x)}{\sin(k\,L)}$

Так не получается, поэтому задаём волновую функцию «вручную» и находим выражение энергии уровня от квантового числа *n*.

```
(%i7) psi(x):=A*sin(nx*%pi*x/L)$
(%i8) (kx^2+ky^2+kz^2)*(h/2/%pi)^2/2/m$
(%i9) subst([kx=nx*%pi/L,ky=ny*%pi/L,kz=nz*%pi/L],%),factor;
```

(%o9) $\dfrac{h^2\,(nz^2+ny^2+nx^2)}{8\,m\,L^2}$

```
(%i10) subst([nz^2+ny^2+nx^2=n^2,L=V^(1/3)],%);
```

(%o10) $\dfrac{h^2\,n^2}{8\,m\,V^{2/3}}$

Находим коэффициент пропорциональности между ΔΩ(ε) и Δε.

```
(%i11) solve(%=epsilon,n);
```

(%o11) $[\,n = -\dfrac{2^{3/2}\sqrt{\varepsilon\,m}\,V^{1/3}}{h},\ n = \dfrac{2^{3/2}\sqrt{\varepsilon\,m}\,V^{1/3}}{h}\,]$

```
(%i12) rhs(second(%));
```

(%o12) $\dfrac{2^{3/2}\sqrt{\varepsilon\,m}\,V^{1/3}}{h}$

```
(%i13) diff(%,epsilon);
```

(%o13) $\dfrac{\sqrt{2}\,m\,V^{1/3}}{h\sqrt{\varepsilon\,m}}$

```
(%i14) 4*%pi*%th(2)^2*%/8,radcan;
```

(%o14) $\dfrac{2^{5/2}\,\pi\sqrt{\varepsilon}\,m^{3/2}\,V}{h^3}$

Нормируем волновую функцию, а также вычисляем среднее значение импульса частицы.

```
(%i15) integrate(psi(x)^2,x,0,L);
```

(%o15) $\frac{A^2 L}{2}$

```
(%i16) solve(%=1,A);
```

(%o16) $[A=-\frac{\sqrt{2}}{\sqrt{L}}, A=\frac{\sqrt{2}}{\sqrt{L}}]$

```
(%i17) integrate(psi(x)*diff(psi(x),x),x,0,L);
```

(%o17) 0

Статистика Ферми-Дирака:

```
(%i18) ksi:1+exp((mu-epsilon)/(k*T))$
(%i19) Ni:k*T*diff(log(ksi),mu);
```

(%o19) $\frac{\%e^{\frac{\mu-\varepsilon}{k\,T}}}{\%e^{\frac{\mu-\varepsilon}{k\,T}}+1}$

```
(%i20) partfrac(%, %e^((mu-epsilon)/(k*T)));
```

(%o20) $1-\frac{1}{\%e^{\frac{\mu-\varepsilon}{k\,T}}+1}$

Статистика Бозе-Эйнштейна:

```
(%i21) ksi:sum(exp(N*(mu-epsilon)/(k*T)),N,0,inf);
```

(%o21) $$\sum_{N=0}^{\infty} \%e^{\frac{(\mu-\varepsilon)\,N}{k\,T}}$$

Отвечаем n (negative)

```
(%i22) %,simpsum;
```

Is $\%e^{\frac{\mu}{k\,T}}-\%e^{\frac{\varepsilon}{k\,T}}$ *positive, negative or zero?*n;

(%o22) $$\frac{1}{1-\%e^{\frac{\mu}{k\,T}-\frac{\varepsilon}{k\,T}}}$$

```
(%i23) Ni:k*T*diff(log(%),mu);
```

(%o23) $$\frac{\%e^{\frac{\mu}{k\,T}-\frac{\varepsilon}{k\,T}}}{1-\%e^{\frac{\mu}{k\,T}-\frac{\varepsilon}{k\,T}}}$$

```
(%i24) partfrac(%, %e^(mu/(k*T)-epsilon/(k*T)));
```

(%o24) $$\frac{\%e^{\frac{\mu}{k\,T}}}{\%e^{\frac{\varepsilon}{k\,T}}-\%e^{\frac{\mu}{k\,T}}}$$

Литература

1. Варикаш В.М., Болсун А.И., Аксёнов В.В. Сборник задач по статистической физике. – 2-е изд., испр. — М.: Эдиториал УРСС, 2004 . — 224 с.

2. Лабовиц Л., Аренс Дж. Задачи по физической химии с решениями / пер. с англ. — М.: Мир, 1972. – 222 с.

3. Левич В.Г. Введение в статистическую физику. — 2- е изд., перераб. — М.: ГИТТЛ, 1954. — 528 с.

4. Кленин В.И., Федусенко И.В. Высокомолекулярные соединения. Учебник. — Саратов: Изд-во СГУ, 2008.

5. Савельев И.В. Курс общей физики. Том 1: механика, колебания и волны, молекулярная физика. — М.: Наука, 1970. — 511 с.

6. Капица П. Л. Маятник с вибрирующим подвесом // УФН. — 1951. — Т. 44, вып. 1. — С. 7–20.

7. Казанский В.Б. Статистическая физика и термодинамика. Задачи, основные понятия и положения: Методическое пособие.– Харьков: ХНУ, 2004.– 112 с.

8. Учебно-методическое пособие по курсу «Термодинамика и статистическая физика». Часть I. Теория вероятностей, термодинамика, классическая статистическая физика / Сост.: Копытин И.В., Алмалиев А.Н., Чуракова Т.А. — Воронеж: ВГУ, 2003. — 59 с. URL: http://window.edu.ru/resource/461/27461/files/apr04107.pdf

9. Краснопевцев Е. А. Статистическая физика равновесных систем (в приложении к микро- и наносистемам) : учеб. пособие — Новосибирск: НГТУ, 2007. — 276 с.

10. Розман Г.А. Задачи по теоретической физике с решениями. ПГПУ, 2007.

11. Лихтенберг А. Динамика частиц в фазовом пространстве. М.: Атомиздат, 1972. — 304 с.

Оглавление